FOR ALL MODELS

THE ORIC PROGRAM BOOK

VINCE APPS

50 PROGRAMS

THE

ORIC-1

PROGRAM BOOK

THE ORIC-1 PROGRAM BOOK

50 programs
for home, educational
and business use

VINCE APPS

PHOENIX PUBLISHING ASSOCIATES
BUSHEY, HERTS.

First published in Great Britain in 1983 by
PHOENIX PUBLISHING ASSOCIATES,
14 Vernon Road, Bushey, Herts. WD2 2JL

ISBN 0 9465 7601 7

Printed in Great Britain by
Billing & Sons Limited, Worcester
Cover design by
Denis Gibney Graphics
Typesetting by
Prestige Press (UK) Ltd

Foreword

If the **Oric-1** is the first computer you have owned then you have made an extremely good choice as you now possess a very versatile machine.

The Oric-1, with colour, full graphics and sound, is a home user's dream and the variety of programs in this book have been written to show off the capabilities at your fingertips.

All of the programs can be used on the 16K, 32K and 48K versions of the **Oric-1** and the necessary instructions to alter listings are given where appropriate.

In addition to providing you with a wide selection of games, educational and business programs, I have included details for budding programmers. You will find how you can redefine characters, input machine codes, use a screen print and peek into the machine memory.

If you have experience in programming you may well understand these functions already but, if you are a newcomer to computing you will find these programs an excellent starting point from which to begin writing your own material.

Many of the programs include suggestions for adapting the content to input your own ideas and improve your skills.

I really feel that this book contains something for everyone and hope that it provides you with hours of entertainment and challenge.

Contents

PROGRAMS FOR FUN AND GAMES

PROGRAMS TO MAKE YOU THINK

PROGRAMS FOR BUDDING PROGRAMMERS

PROGRAMS FOR FUN AND GAMES

This first selection of programs is designed to show off the colour, sound, graphics and speed of your **Oric.**

REACTION

Here is a simple little test to check your reaction speed before you try some of the tougher space games.

The Oric will emit a tone – hit any key to stop the counter. Beat the clock.

If you are having a party and think some people shouldn't drive home let them try this test, it could be a lot cheaper than a breathalyser test.

```
100 REM
110 REM  REACTION
120 REM
130 CLS
140 PRINT:PRINT
150 IF RND(1)>.995 THEN 200
160 K$=KEY$
170 IF K$="" THEN 150
180 PRINT "CHEAT!!"
190 GOTO 150
200 MUSIC 1,3,6,0
210 PLAY 1,0,1,2000
220 K$=KEY$
230 IF K$="" THEN C=C+1:GOTO 220
240 PRINT "YOUR REACTION TIME IS ";C
250 PRINT
260 C=0
270 GOTO 150
```

CATERPILLAR

You are in control of the body of a fast growing caterpillar. Move using the O and P keys for left and right and the Q and A keys for up and down. These control keys can be changed, if desired, by altering lines 290-320. Scattered around the screen are a number of spiders which must be avoided. If you hit either your own body, or one of the spiders, or try to leave the screen, the game ends. One point is scored for each body segment produced. Score 320 and a new screen will be generated.

There are two routines of particular interest in this program, one is the delay loop in line 370. This has the effect of speeding up the movement as the score increases. The second is the "High Score" table in line 540 onwards. This routine can be incorporated in many of the scoring games in this book.

The memory locations in line 140 are for the 48K Oric. If you have the 32K machine, use 30656 and 30671. For the 16K version use 14272 and 14287.

```
100 REM
110 REM  CATERPILLAR
120 REM
130 CLS
135 REM  DEFINE CHARACTERS
140 FOR J=47040 TO 47055
150 READ A:POKE J,A
160 NEXT J
170 DATA 8,28,30,63,62,28,8,0,33,18,44
,31,30,45,18,33
175 REM  SET UP SCREEN
180 INK 2:PAPER 0
190 FOR J=1 TO 30
200 R1=INT(RND(1)*37)+1
210 R2=INT(RND(1)*25)+1
220 PLOT R1,R2,121
230 NEXT J
240 X=19:Y=13:A$="A"
```

```
250 PRINT CHR$(17)
255 REM  LOOK AT KEYBOARD
260 T$=KEY$
270 IF T$="" THEN 290
280 A$=T$
290 IF A$="Q" THEN Y=Y-1
300 IF A$="A" THEN Y=Y+1
310 IF A$="O" THEN X=X-1
320 IF A$="P" THEN X=X+1
325 REM  TEST FOR SCREEN EDGE
330 IF X>38 OR X<0 OR Y>26 OR Y<0 THEN
 500
332 IF SCRN(X,Y)<>32 THEN 500
340 MUSIC 1,3,5,1
350 PLAY 1,0,1,400
360 SC=SC+1
365 REM  DELAY LOOP
370 FOR D=1 TO 250-SC:NEXT D
375 REM  NEW SCREEN
380 IF SC=320 THEN CLS:GOTO 190
400 PLOT X,Y,120
410 GOTO 260
490 REM  HIT
500 PAPER 1
510 EXPLODE
520 PAPER 0
530 FOR J=1 TO 26:PRINT:NEXT J
540 PRINT "YOUR SCORE: ";SC
550 PRINT CHR$(17)
560 INPUT "YOUR NAME";NM$
570 IF LEN(NM$)>12 THEN PRINT "TOO LON
G!":GOTO 560
575 REM  BUBBLE SORT
580 S(10)=SC
590 N$(10)=NM$
600 FOR J=10 TO 2 STEP-1
610 IF S(J)<S(J-1) THEN 680
620 TP=S(J-1)
630 T$=N$(J-1)
640 S(J-1)=S(J)
650 N$(J-1)=N$(J)
```

```
660 S(J)=TP
670 N$(J)=T$
680 NEXT J
690 CLS:INK 0:PAPER 7:PRINT
700 PRINT "           HIGH SCORES":PRIN
T
710 FOR M=1 TO 9
720 PRINT " ";M;" ";S(M);"    ";
730 PRINT N$(M)
740 NEXT M
750 PRINT:PRINT "         PRESS P TO PL
AY"
760 PRINT "               S TO STOP"
765 REM  LOOK AT KEYBOARD
770 GET Q$
780 IF Q$="S" THEN 810
790 SC=0:CLS
800 GOTO 180
810 END
```

MISSILE DEFENCE

You have six cities to protect from an avalanche of enemy missiles. Move your laser sight with the four arrow keys. Fire with the space bar.

The game will end if all six cities are destroyed.

The memory locations used in line 200 for character definition are for the 48K Oric. If you have the 32K version these should be replaced with 30672 and 30703. For the 16K machine use 14288 and 14319.

```
100 REM
110 REM  MISSILE DEFENCE
120 REM
130 CLS
135 REM  DISABLE CURSOR
140 PRINT CHR$(17)
170 REM  PRINT TITLE PAGE
180 PLOT 11,12,"MISSILE DEFENCE"
190 WAIT 200
192 CLS
193 REM  MOVE CURSOR POSITION
194 FOR M=1 TO 26
196 PRINT
198 NEXT M
199 REM  DEFINE CHARACTERS
200 FOR J=47056 TO 47087
210 READ A
220 POKE J,A
230 NEXT J
240 DATA 63,45,63,45,63,45,63,63
250 DATA 0,28,28,28,28,28,28,8
260 DATA 12,12,12,63,63,12,12,12
270 DATA 42,21,42,21,42,21,42,21
280 REM  INITIAL VALUES
290 SX=18:SY=15
300 FOR J=1 TO 9
310 R(J)=2
320 C(J)=INT(RND(1)*28)+4
```

```
330 NEXT J
340 GOSUB 1000
350 J=0:BL=18:ML=75
360 D$="SCORE: "+STR$(SC)
370 PLOT 21,0,D$
380 PLOT 0,0,2
390 D$="MISSILES: "+STR$(ML)
392 PLOT 4,0,D$
400 GOTO 1200
410 REM  MOVE MISSILES
420 J=J+1
430 IF J=10 THEN J=1
440 IF J>1+SC/20 THEN RETURN
460 PLOT C(J),R(J)," "
470 R(J)=R(J)+2
480 X=SCRN(C(J),R(J))
490 IF X<>122 THEN 600
500 PAPER 1
510 EXPLODE
520 PLOT C(J),R(J),")"
530 WAIT 20
540 PAPER 7
550 PLOT C(J),R(J)," "
552 PLOT C(J),R(J)+1," "
554 PLOT C(J),R(J)+2," "
560 BL=BL-1
570 IF BL=0 THEN 1300
580 GOTO 610
600 IF R(J)<>26 THEN 630
610 R(J)=2
620 C(J)=INT(RND(1)*28)+4
622 IF SCRN(C(J),26)=32 THEN 620
630 PLOT C(J),R(J),"("
640 RETURN
700 REM  MOVE SIGHTS
710 K$=KEY$
720 IF K$="" THEN 740
730 K=ASC(K$)
740 OX=SX
750 OY=SY
760 IF K=11 AND SY>1 THEN SY=SY-1
```

```
770 IF K=10 AND SY<23 THEN SY=SY+1
780 IF K=9 AND SX<37 THEN SX=SX+1
790 IF K=8 AND SX>1 THEN SX=SX-1
794 IF ML<1 THEN 940
798 REM  SHOOT
800 IF K<>32 THEN 940
802 K=0
810 ZAP
812 ML=ML-1
814 D$=STR$(ML)+" "
816 PLOT 14,0,D$
818 REM  TEST FOR HIT
820 FOR M=1 TO 9
822 IF C(M)=SX AND R(M)=SY THEN 830
826 NEXT M
828 GOTO 940
829 REM  HIT
830 PAPER 1
840 EXPLODE
850 SC=SC+10
860 D$=STR$(SC)
870 PLOT 28,0,D$
880 PAPER 7
910 C(M)=INT(RND(1)*28)+4
912 IF SCRN(C(M),26)=32 THEN 910
920 R(M)=2
940 PLOT OX,OY," "
942 LA=SCRN(SX,SY)
950 PLOT SX,SY,"¦"
980 RETURN
995 REM  BUILD CITIES
1000 FOR N=4 TO 29 STEP 5
1010 FOR P=24 TO 26
1020 PLOT N,P,"zzz"
1100 NEXT P,N
1110 PLOT SX,SY,"¦"
1120 RETURN
1190 REM  MAIN LOOP
1200 GOSUB 420
1210 GOSUB 710
1220 GOTO 1200
```

```
1300 IF HS<SC THEN HS=SC
1320 PRINT
1330 PRINT "HIGH SCORE: ";HS
1340 PRINT
1350 PRINT "YOUR SCORE: ";SC
1360 PRINT
1370 INPUT "PLAY AGAIN";Q$
1380 IF LEFT$(Q$,1)="N" THEN 1410
1385 REM  RESET
1390 SC=0:CLS
1392 RESTORE
1400 GOTO 194
1410 PRINT CHR$(17)
1420 END
```

OBSTACLE COURSE

The object of this game is very simple.

All the player has to do is to find his way from the top to the bottom of the screen using the four arrow keys. There are a few snags however. The player has to complete the game within 50 moves, avoid the obstacles and guide themselves by sound alone!!

The memory locations in line 150 are for the 48K Oric. For the 32K machine use 30648 and 30655. For the 16K use 14264 and 14271.

```
100 REM
110 REM  OBSTACLE COURSE
120 REM
130 CLS
132 PRINT:PRINT
134 INPUT "SKILL LEVEL (1 TO 5)";SK
135 IF SK<1 OR SK>5 THEN 134
136 SK=SK/10
138 CLS
140 PRINT CHR$(17)
145 REM  DEFINE CHARACTER
150 FOR J=47032 TO 47039
160 READ A
170 POKE J,A
180 NEXT J
190 DATA 63,63,63,63,63,63,63,63
195 REM  INITIAL VALUES
200 MV=50
220 C=18:R=26
225 REM  PLOT OBSTACLES
230 FOR J=2 TO 25
240 FOR K=1 TO 38
250 IF RND(1)>SK THEN 270
260 PLOT K,J,119
270 NEXT K,J
280 WAIT 200
290 FOR J=1 TO 20
```

```
300 PLOT C,R,"X"
310 WAIT 25
320 PLOT C,R," "
322 WAIT 25
330 NEXT J
340 PAPER 0:INK 0
345 REM  MAIN LOOP
350 D$="MOVES LEFT = "+STR$(MV)+" "
360 PLOT 10,0,D$
362 PLOT 0,0,2
365 REM  LOOK AT KEYBOARD
370 K$=KEY$
372 IF K$="" THEN 370
376 MV=MV-1
378 IF MV=0 THEN 550
380 CO=C
390 RO=R
400 IF K$=CHR$(8) THEN C=C-1
410 IF K$=CHR$(9) THEN C=C+1
420 IF K$=CHR$(10) THEN R=R+1
430 IF K$=CHR$(11) THEN R=R-1
440 IF R>26 OR C>37 OR C<2 THEN 500
450 IF R<2 THEN 670
460 IF SCRN(C,R)=119 THEN 500
470 MUSIC 1,3,6,0
480 PLAY 1,0,1,1000
490 GOTO 350
500 MUSIC 1,1,3,0
510 PLAY 1,0,1,1000
520 C=CO
530 R=RO
540 GOTO 350
545 REM  FAILURE
550 FOR J=12 TO 1 STEP-1
560 MUSIC 1,2,J,0
570 PLAY 1,0,1,1000
580 WAIT 25
590 NEXT J
600 PLOT C,R,"X"
610 PAPER 7:INK 0
620 FOR M=1 TO 22
```

```
630 PRINT
640 NEXT M
650 PRINT CHR$(17)
660 GOTO 730
665 REM  SUCCESS
670 FOR J=1 TO 12
680 MUSIC 1,3,J,0
690 PLAY 1,0,1,1000
700 WAIT 25
710 NEXT J
720 GOTO 600
730 END
```

ALIEN STORM

In this game you are pitted against a never-ending stream of bomb dropping aliens descending from the top of the screen. Move your laser base using the O and P keys. Fire with the space-bar. If you are hit by a bomb, or an alien lands on your base, the game ends.

The memory locations in line 810 are for the 48K Oric. For the 32K model use 30664 and 30695. For the 16K version use 14280 and 14311.

```
100 REM
110 REM  ALIEN STORM
120 REM
130 CLS
150 GOTO 810
190 REM MOVE ALIEN
200 PLOT AX,AY," "
210 AY=AY+1
220 IF AY=27 THEN AY=1:AX=INT(RND(1)*3
6)+1
230 IF AY=26 AND BX=AX THEN 1010
235 REM  WHICH TYPE
240 IF AY/2=INT(AY/2) THEN AC$=CHR$(12
1) ELSE AC$=CHR$(122)
250 PLOT AX,AY,AC$
260 MUSIC 1,3,5,0
270 PLAY 1,0,1,200
300 RETURN
310 REM MOVE BOMB
320 IF M=0 THEN M=1:MX=AX:MY=AY:GOTO 4
00
330 PLOT MX,MY," "
340 MY=MY+2
345 REM  HIT
350 IF MY>25 AND MX=BX THEN 1010
360 IF MY>25 THEN M=0:GOTO 320
370 PLOT MX,MY,":"
400 RETURN
```

```
410 REM  MOVE BASE
420 T$=KEY$
430 IF T$=""THEN 450
440 A$=T$
450 PLOT BX,BY," "
460 IF A$="O" AND BX>1 THEN BX=BX-1
470 IF A$="P" AND BX<37 THEN BX=BX+1
475 REM  FIRE LASER
480 IF A$<>" " THEN 630
490 PLOT BX,BY-1,S$
500 ZAP
502 PLOT BX,BY-1," "
504 A$="P"
510 IF BX<>AX THEN 630
515 REM  ON TARGET
520 PLOT BX,BY,B$
530 PAPER 1
540 EXPLODE
550 SC=SC+10
560 D$=STR$(SC)
570 D1$=RIGHT$(D$,LEN(D$)-1)
572 WAIT 5
574 PLOT BX,BY-1," "
580 PLOT 19,0,D1$
590 PAPER 7
592 PLOT AX,AY," "
595 REM  NEW ALIEN
600 AY=1
610 AX=INT(RND(1)*36)+1
630 PLOT BX,BY,B$
700 RETURN
800 REM  DEFINE CHARACTERS
810 FOR J=47048 TO 47079
820 READ A:POKE J,A
830 NEXT J
840 DATA 0,30,63,45,63,12,18,33,0,30,6
3,45,63,12,12,12
850 DATA 0,12,12,30,63,63,63,63,12,12,
12,12,12,12,12,12
855 REM  ASSIGN CHARACTERS
860 S$="¦"
```

```
870 B$="("
880 BY=26:BX=10:AY=1:AX=15
890 A$="P"
900 FOR J=1 TO 26:PRINT:NEXT J
910 PRINT CHR$(17)
920 PLOT 12,0,"SCORE: "
925 REM  MAIN LOOP
950 GOSUB 200
954 GOSUB 420
960 GOSUB 310
970 GOSUB 420
990 GOTO 950
1000 REM  BASE HIT
1010 PAPER 1
1020 EXPLODE
1030 PLOT 10,12,"G A M E    O V E R"
1032 PAPER 7
1040 PRINT "YOUR SCORE: ";SC
1042 PRINT CHR$(17)
1050 END
```

MONEYBAGS

It's pennies from heaven, or at least dollar signs from the top of the screen, in this game for your **Oric** You control a catcher at the bottom of the screen, moving left and right with the O and P keys. Catch as many of the dollar signs as you can, but miss 20 and the game ends.

```
100 REM
105 REM  MONEYBAGS
110 REM
116 CLS
120 PLOT 10,12,"M O N E Y B A G S"
122 WAIT 500
124 GOTO 500
125 REM  MOVE DOLLARS
130 J=J+1
132 IF J=10 THEN J=1
134 IF J>1+SC/20 THEN 330
136 IF RND(1)>.9 THEN 330
140 PLOT C(J),R(J)," "
150 R(J)=R(J)+1
155 REM  CATCH?
160 IF R(J)<>25 THEN 220
170 IF C(J)<>CC THEN 220
180 MUSIC 1,4,5,0
190 PLAY 1,0,1,350
195 REM  INCREMENT SCORE
200 SC=SC+10
210 GOTO 270
215 REM  MISSED
220 IF R(J)<>26 THEN 300
230 MUSIC 1,3,3,0
240 PLAY 1,0,1,400
250 CN=CN+1
255 REM  END OF GAME
260 IF CN=20 THEN 700
265 REM  NEW POSITION
270 C(J)=INT(RND(1)*33+2)
```

```
280 R(J)=1
290 GOTO 330
300 PLOT C(J),R(J),36
330 RETURN
395 REM  MOVE CATCHER
400 K$=KEY$
410 IF K$="" THEN 460
420 PLOT CC,CR," "
430 IF K$="O" AND CC>2 THEN CC=CC-1
440 IF K$="P" AND CC<36 THEN CC=CC+1
450 PLOT CC,CR,"-"
460 RETURN
490 REM  START
500 CLS
502 PAPER 0
504 INK 7
506 J=0
507 CN=0
508 SC=0
510 PRINT CHR$(17)
520 FOR M=1 TO 26
530 PRINT
540 NEXT M
550 CR=25:CC=18
560 PLOT CC,CR,"-"
570 FOR M=1 TO 9
580 R(M)=1
590 C(M)=INT(RND(1)*32+1)
600 NEXT M
605 REM  MAIN LOOP
610 GOSUB 400
620 GOSUB 130
625 REM  PRINT SCORE
630 D$="SCORE: "+STR$(SC)
640 PLOT 14,0,D$
650 GOTO 610
700 FOR M=10 TO 1 STEP -1
710 MUSIC 1,3,M,0
720 PLAY 1,0,1,400
722 WAIT 20
730 NEXT M
```

```
740 PLOT 11,12,"G A M E   O V E R"
750 WAIT 200
760 PRINT
762 IF HS<SC THEN HS=SC
764 PRINT "HIGH SCORE = ";HS
766 PRINT
770 PRINT "YOUR SCORE = ";SC
780 PRINT CHR$(17)
790 INPUT "PLAY AGAIN";Q$
800 IF LEFT$(Q$,1)="N" THEN 810 ELSE
 500
810 PAPER 7:INK 0
820 END
```

ELECTRONIC ORGAN

This program turns your versatile **Oric** into a mini electronic organ!

The top letter row (starting with Q) forms the keyboard. There is a choice of envelope and note length to give different effects.

Try starting with envelope 1 and a note length of 2000. As it stands, it only covers 12 notes, but this range can easily be extended, following the format of lines 185 to 240, using a second row of keys.

```
100 REM
105 REM  ELECTRONIC ORGAN
110 REM
115 CLS
120 PRINT
125 INPUT "ENVELOPE (1-7)";E
130 IF E<1 OR E>7 THEN 125
135 PRINT
140 INPUT "NOTE LENGTH (1-32767)";L
145 IF L<1 OR L>32767 THEN 140
150 PRINT
155 PRINT "PRESS QWERTYUIOP[ ] TO PLAY"

160 PRINT
165 PRINT "PRESS Z TO STOP"
170 PRINT
175 GET A$
180 IF A$="Z" THEN 270
185 IF A$="Q" THEN O=3:N=1
190 IF A$="W" THEN O=3:N=3
195 IF A$="E" THEN O=3:N=5
200 IF A$="R" THEN O=3:N=6
205 IF A$="T" THEN O=3:N=8
210 IF A$="Y" THEN O=3:N=10
215 IF A$="U" THEN O=3:N=12
220 IF A$="I" THEN O=4:N=1
225 IF A$="O" THEN O=4:N=3
```

```
230 IF A$="P" THEN O=4:N=5
235 IF A$="[" THEN O=4:N=6
240 IF A$="]" THEN O=4:N=8
245 IF O=0 OR N=0 THEN 175
250 MUSIC 1,O,N,0
255 PLAY 1,0,E,L
260 N=0:O=0
265 GOTO 175
270 END
```

LUNAR LANDER

Now's your chance to start your astronaut training as you endeavour to guide your lunar module to a safe landing on the moon's surface.

Control the module's rate of descent by pressing the space-bar to fire your retro-rockets, but watch your fuel consumption. As it stands the game is difficult, but not impossible, to master. For the faint-hearted, the amount of fuel in line 230 can be increased to make the game easier.

The memory locations in line 160 are for the 48K machine. If you have the 32K version, use 30672 and 30687. For the 16K machine use 14288 and 14303.

```
100 REM
110 REM  LUNAR LANDER
120 REM
130 CLS
140 PAPER 0:INK 7
150 REM DEFINE CHARACTERS
160 FOR J=47056 TO 47071
170 READ A
180 POKE J,A
190 NEXT J
200 DATA
208 DATA 0,30,63,63,63,63,18,51
210 DATA 63,63,63,63,63,63,63,63
215 REM  INITIAL VALUES
220 HT=2
230 VEL=0:FU=40
240 PRINT CHR$(17)
250 REM  PLOT LUNAR SURFACE
260 PLOT 1,26,"<<<<<<<<<<<<<<<<<<<<<<<<<<<<
<<<<<<<<<<<<<<<<"
270 PLOT 1,25,"<<<<"
280 PLOT 35,25,"<<<<"
290 PLOT 1,24,"<<<"
300 PLOT 36,24,"<<<"
```

```
310 REM  MAIN LOOP
318 IF HT<1 THEN HT=1
320 PLOT 18,HT,122
322 MUSIC 1,2,12,0
324 PLAY 1,1,1,600
326 WAIT 50
330 VEL=VEL+.5
340 M$="FUEL = "+STR$(FU)
350 PLOT 14,0,M$
360 PLOT 0,0,2
370 IF FU=0 THEN 510
400 REM  IGNITE?
410 K$=KEY$
420 IF K$="" THEN 510
422 IF K$="S" THEN 710
430 SHOOT
440 WAIT 10
450 REM  DECREMENT FUEL
460 FU=FU-10
470 IF FU>0 THEN 490
480 FU=0
490 VEL=VEL-1
510 PLOT 18,HT," "
520 HT=INT(HT+VEL)
530 IF HT>=25 THEN 560
540 GOTO 318
550 REM CONTACT
560 IF VEL<2.1 THEN 610
570 EXPLODE
580 PLOT 18,26,122
590 PLOT 13,12,"YOU CRASHED!"
600 GOTO 670
610 MUSIC 1,3,6,0
620 PLAY 1,0,1,2000
630 WAIT 50
640 MUSIC 1,3,12,0
650 PLAY 1,0,1,3000
652 WAIT 60
654 MUSIC 1,4,3,0
656 PLAY 1,0,1,4000
658 WAIT 70
```

```
660 PLOT 11,12,"A PERFECT LANDING!"
670 WAIT 500
680 RESTORE
690 PRINT CHR$(17)
700 GOTO 130
710 PRINT CHR$(17)
720 CLS
730 PAPER 7
740 INK 0
750 END
```

ANAGRAM

This game will print a series of anagrams for you to decipher. If your answer is correct, a new anagram will be given. Type "QUIT" to give up on a particular anagram and "STOP" to end the game.

The list of words in the data statements of lines 520 to 540 can of course be extended or changed to suit individual preferences.

```
100 REM
110 REM  ANAGRAM
120 REM
130 CLS
140 PRINT:PRINT
150 DIM WD$(100),M$(12)
160 J=1
170 REM  READ IN DATA
180 READ WD$(J)
190 IF WD$(J)="END" THEN 220
200 J=J+1
210 GOTO 180
220 J=J-1
230 REM  SELECT WORD
240 R=INT(RND(1)*J)+1
250 A$=WD$(R)
260 FOR I=1 TO LEN(A$)
270 R=INT(RND(1)*12)+1
280 IF M$(R)<>"" THEN 270
290 M$(R)=MID$(A$,I,1)
300 NEXT I
310 FOR I=1 TO 12
320 IF M$(I)="" THEN 340
330 PRINT M$(I);
340 NEXT I
350 PRINT:PRINT
360 INPUT "YOUR GUESS";G$
370 PRINT
```

```
380 IF G$=A$ THEN 430
390 IF G$="QUIT" THEN PRINT A$:PRINT:G
OTO 450
400 IF G$="STOP" THEN END
410 PRINT "TRY AGAIN!":PRINT
420 GOTO 360
430 PRINT "CORRECT!"
440 NC=NC+1
450 A$=""
460 FOR I=1 TO 12
470 M$(I)=""
480 NEXT I
490 PRINT
500 GOTO 240
520 DATA POND,WOOD,MOUSE,TIGER,HOUSE,A
PPLE,TRAIN,WORD,PARTY,CHIMNEY
530 DATA DIGIT,IDEA,ANAGRAM,ORBIT,PENN
Y,PEAR,BINARY,PUPIL
540 DATA VIDEO,RECORD,MODULAR,LEAF,BOO
K,FILM,CRICKET,PICTURE
550 DATA END
```

BREAKOUT

This is a version of the popular arcade game of the same name. There are five rows of coloured bricks at the top of the screen. Try and demolish them with the bouncing ball from your bat. Move your bat with the O and P keys.

A point of interest in the program is the use of the plot command in lines 280 – 320 to give different coloured rows of bricks. By plotting a number between 0 and 7 in column 0 in any row, the ink value will take the colour of that number. This can be used to good effect in many games.

The memory locations in line 160 are for the 48K **Oric** For the 32K version use 30672 and 30695. For the 16K, use 14288 and 14311.

```
100 REM
110 REM  BREAKOUT
120 REM
130 CLS
140 PAPER 0:INK 7
142 PRINT CHR$(17)
150 REM  DEFINE CHARACTERS
160 FOR J=47056 TO 47079
170 READ A
180 POKE J,A
190 NEXT J
200 DATA 0,30,30,30,30,30,30,0
210 DATA 0,0,0,0,0,0,63,63
220 DATA 0,0,30,30,30,0,0,0
230 REM  SET UP WALL
240 FOR J=3 TO 7
250 FOR K=2 TO 36
260 PLOT K,J,122
270 NEXT K,J
280 PLOT 0,3,1
290 PLOT 0,4,2
300 PLOT 0,5,3
310 PLOT 0,6,4
320 PLOT 0,7,5
```

```
325 REM  INITIAL VALUES
330 BL=5
340 BC=16:BR=24
350 C=16:X=-1:Y=-1
360 R=INT(RND(1)*10)+10
380 GOSUB 900
390 D$="BALLS: "+STR$(BL)
400 PLOT 8,0,D$
410 D$="SCORE: "+STR$(SC)
420 PLOT 20,0,D$
422 PLOT 0,0,2
430 PLOT C,R," "
440 C=C+X
450 R=R+Y
452 IF C>37 OR C<1 OR R>26 OR R<1 THEN
 460
454 LA=SCRN(C,R)
460 IF C<2 OR C>36 THEN X=-X
470 IF R<2 THEN Y=-Y
480 IF R<26 THEN 590
490 MUSIC 1,1,9,0
500 PLAY 1,0,1,200
510 BL=BL-1
512 D$=STR$(BL)
514 PLOT 15,0,D$
520 IF BL=0 THEN 1000
540 GOSUB 950
560 PLOT C,R," "
570 GOTO 340
590 IF LA=32 THEN 670
600 IF LA<>123 THEN 620
604 Y=-Y:X=-X
606 MUSIC 1,4,2,0
608 PLAY 1,0,1,200
610 IF RND(1)>.5 THEN X=-X
614 GOTO 700
620 IF LA=122 THEN Y=-Y
630 PLOT C,R," "
640 MUSIC 1,3,7,0
642 SC=SC+10
650 PLAY 1,0,1,200
```

```
670 PLOT C,R,"¦"
700 K$=KEY$
704 GOSUB 900
710 IF K$="" THEN 790
720 GOSUB 950
750 IF K$="O" AND BC>1 THEN BC=BC-2
760 IF K$="P" AND BC<35 THEN BC=BC+2
780 GOSUB 900
790 D$=STR$(SC)
800 PLOT 27,0,D$
810 GOTO 430
895 REM  BAT
900 FOR M=0 TO 2
910 PLOT BC+M,BR,"("
920 NEXT M
930 RETURN
945 REM  NO BAT
950 FOR M=0 TO 2
960 PLOT BC+M,BR," "
970 NEXT M
980 RETURN
1000 IF SC>HS THEN HS=SC
1010 FOR M=1 TO 27:PRINT:NEXT M
1020 PRINT "HIGH SCORE: ";HS
1030 PRINT
1040 PRINT "YOUR SCORE: ";SC
1050 PRINT
1060 INPUT "PLAY AGAIN";Q$
1070 IF LEFT$(Q$,1)="N" THEN 1110
1080 SC=0
1090 CLS
1100 GOTO 230
1110 PRINT CHR$(17)
1120 PAPER 7:INK 0
1130 END
```

RESCUE

Ever fancied being a knight in shining armour?

Here's your chance as you strive to rescue the maiden in distress from the black magician's castle.

Move your knight using keys O and P for left and right, and Q and A for up and down.

One problem – the evil magician is throwing large boulders at you! If you are hit by, or bump into one, you fail.

The difficulty of the game can be varied by changing the loop count in line 630.

The memory locations in line 560 are for the 48K **Oric** If you have the 32K machine these should be changed to 30672 and 30703. For the 16K version use 14288 and 14319.

```
100 REM
105 REM  RESCUE
110 REM
115 CLS
120 PLOT 12,12,"R E S C U E"
125 WAIT 500
130 GOTO 500
140 REM  MOVE
150 K$=KEY$
160 IF K$="" THEN RETURN
170 PLOT C,R," "
180 IF K$="O" AND C>1 THEN C=C-1
190 IF K$="P" AND C<37 THEN C=C+1
200 IF K$="Q" AND R>1 THEN R=R-1
210 IF K$="A" AND R<26 THEN R=R+1
220 IF SCRN(C,R)=32 THEN 280
230 IF SCRN(C,R)=122 OR SCRN(C,R)=123
THEN 800
235 REM  HIT BY ROCK
240 EXPLODE
250 PLOT 12,12,"S P L A T !"
260 WAIT 100
```

```
270 GOTO 901
280 PLOT C,R,"|"
295 REM  THROW ROCK
300 R1=INT(RND(1)*12)+R-6
310 IF R1>26 OR R1<1 THEN 300
320 R2=INT(RND(1)*12)+C-6
330 IF R2>36 OR R2<1 THEN 320
332 IF RND(1)>.5 THEN RETURN
340 X=SCRN(R2,R1)
345 REM  CASTLE
350 IF X>121 AND X<124 THEN RETURN
360 IF X=124 THEN PLOT R2,R1,")":GOTO
240
370 MUSIC 1,2,3,0
380 PLAY 1,0,1,300
390 PLOT R2,R1,")"
400 RETURN
495 REM  START
500 CLS
505 REM  TURN OFF CURSOR
510 PRINT CHR$(17)
520 FOR J=1 TO 26
530 PRINT
540 NEXT J
550 PAPER 0:INK 7
555 REM  REDEFINE CHARACTERS
560 FOR J=47056 TO 47087
570 READ A
575 POKE J,A
580 NEXT J
590 DATA 42,63,42,63,42,63,46,62
600 DATA 42,62,42,52,42,62,26,30
610 DATA 12,16,12,31,44,10,18,18
620 DATA 12,30,62,63,63,31,30,12
625 REM  SET UP SCREEN
630 FOR J=1 TO 180
640 R1=INT(RND(1)*35)+1
650 R2=INT(RND(1)*25)+1
660 PLOT R1,R2,")"
670 NEXT J
675 REM  ASSIGN CHARACTERS
```

```
680 PLOT 34,1,"z"
690 PLOT 35,1,"("
700 R=25
705 C=2
710 PLOT C,R,"!"
715 REM  MAIN LOOP
760 GOSUB 295
770 GOSUB 150
780 GOTO 760
790 REM  SUCCESS
800 FOR M=1 TO 4
810 PLAY 0,0,0,0
820 READ N
830 MUSIC 1,3,N,0
840 PLAY 1,0,5,100
850 WAIT 50
860 NEXT M
870 WAIT 10
880 PLAY 0,0,0,0
890 DATA 5,12,2,12
900 PLOT 12,12,"YOU MADE IT!!"
901 PRINT
902 PRINT "YOUR SCORE: ";(C+(26-R))*10

903 PRINT
905 REM  END ROUTINE
910 INPUT "PLAY AGAIN";Q$
912 PRINT CHR$(17)
920 IF LEFT$(Q$,1)="N" THEN 940 ELSE R
UN
940 PAPER 7:INK 0
950 END
```

ARTIST

Using the keyboard of your **Oric** you can draw straight onto the high resolution screen.

As you will see the instruction for control keys is in the program itself so, once you have entered the program, **Oric** will do the rest.

```
100 REM
105 REM  ARTIST
110 REM
120 CLS:PRINT:PRINT
121 REM  PRINT INSTRUCTIONS
122 PRINT "USE KEYS 1 TO 7 TO CHANGE C
OLOUR":PRINT
124 PRINT "ARROW KEYS AND UIJK TO DRAW
":PRINT
126 PRINT "B KEY TO TURN OFF":PRINT
128 PRINT "D KEY TO TURN ON":PRINT
130 PRINT "S KEY TO STOP":PRINT
132 PRINT "PRESS SPACEBAR":GET N$
134 X=120:Y=100
136 HIRES
138 REM  LOOK AT KEYBOARD
140 GET K$
152 IF K$="B" THEN B=1
154 IF K$="D" THEN B=0
156 IF K$="S" THEN TEXT:END
160 IF K$=CHR$(11) AND Y>2 THEN Y=Y-1
170 IF K$=CHR$(10) AND Y<197 THEN Y=Y+
1
180 IF K$=CHR$(8) AND X>2 THEN X=X-1
190 IF K$=CHR$(9) AND X<237 THEN X=X+1

192 IF K$="U" AND X>2 AND Y>2 THEN X=X
-1:Y=Y-1
194 IF K$="I" AND X<237 AND Y>2 THEN X
=X+1:Y=Y-1
```

```
196 IF K$="J" AND X>2 AND Y<197 THEN X
=X-1:Y=Y+1
198 IF K$="K" AND X<237 AND Y<197 THEN
 X=X+1:Y=Y+1
199 REM  CHANGE COLOUR
200 IF ASC(K$)>47 AND ASC(K$)<56 THEN
INK VAL(K$)
210 IF B=1 THEN 230
220 CURSET X,Y,1
224 GOTO 140
230 CURSET X,Y,0
234 GOSUB 500
240 GOTO 140
500 DRAW 1,1,1
505 CURMOV -1,-1,3
510 DRAW 1,1,0
515 CURMOV -1,-1,3
520 RETURN
```

CIRCLES

Your **Oric** is capable of drawing some beautiful and intricate patterns on the high resolution screen.

The next three programs, Circles, Web and Chopsticks, are designed to show some of the effects which are possible. The outstanding point to consider is the effect of these programs from relatively short listings.

All of these programs are simple to adapt to show off your own design skills.

```
100 REM
110 REM  CIRCLES
120 REM
130 HIRES
135 IF RND(1)>.01 THEN 180
140 C=INT(RND(1)*6+1)
150 S=INT(RND(1)*6+1)
155 IF S=C THEN 150
160 PAPER C
170 INK S
180 X=INT(RND(1)*220)+10
190 Y=INT(RND(1)*180)+10
200 R=INT(RND(1)*10)+5
210 IF X+R>238 OR X-R<1 THEN 180
220 IF Y+R>198 OR Y-R<1 THEN 190
230 CURSET X,Y,0
240 CIRCLE R,1
250 GOTO 135
```

WEB

```
100 REM
110 REM  WEB
120 REM
130 FOR M=1 TO 10
140 HIRES
150 R1=INT(RND(1)*6)+1
160 FOR J=1 TO 20
170 A=INT(RND(1)*200)+20
180 B=INT(RND(1)*150)+20
190 R2=INT(RND(1)*6)+1
200 IF R1=R2 THEN 190
210 PAPER R1:INK R2
230 R=1
240 CURSET A,B,3
250 CIRCLE R,1
260 R=R+3
270 IF A+R>235 OR B+R>193 THEN 300
280 IF A-R<9 OR B-R<9 THEN 300
290 GOTO 250
300 NEXT J
310 WAIT 1000
320 NEXT M
330 TEXT
340 END
```

CHOPSTICKS

```
100 REM
110 REM  CHOPSTICKS
120 REM
130 HIRES
132 R=INT(RND(1)*6)+1
134 INK R
140 R1=INT(RND(1)*220)+10
150 R2=INT(RND(1)*180)+10
160 CURSET R1,R2,0
170 R3=INT(RND(1)*100)+10
180 R4=INT(RND(1)*50)+10
190 IF (R1+R3)>238 THEN 140
200 IF (R2+R4)>198 THEN 140
210 DRAW R3,R4,1
220 CN=CN+1
230 IF CN<250 THEN 140
240 WAIT 1000
250 CN=0
260 GOTO 130
```

COMPOSER

Your **Oric** has some very versatile sound effects, not available on most home computers, due to its use of a special sound synthesis "chip".

I've included two programs to illustrate some of its capabilities; Composer, which generates a random series of notes and Clementine, which gives a fair rendering of this well known tune.

```
100 REM
110 REM  COMPOSER
120 REM
130 E=INT(RND(1)*2)+1
140 P=INT(RND(1)*1000)+500
150 O=INT(RND(1)*6)+1
160 N=INT(RND(1)*12)+1
170 MUSIC 1,O,N,0
180 PLAY 1,0,E,P
184 W=INT(RND(1)*20)+5
190 WAIT W
200 CN=CN+1
210 IF CN<250 THEN 130
216 WAIT 50
220 PLAY 0,0,0,0
230 END
```

CLEMENTINE

```
100 REM
110 REM  CLEMENTINE
120 REM
130 READ N
140 IF N<0 THEN 0=-N:GOTO 130
150 IF N=0 THEN 200
160 MUSIC 1,0,N,0
170 PLAY 1,0,1,2000
180 WAIT 45
190 GOTO 130
200 RESTORE
210 WAIT 500
220 GOTO 130
250 DATA -2,8,8,8,3,12,12,12,8,8,12,-3
,3,3,1,-2,12,10,10,12
260 DATA -3,1,1,-2,12,10,12,8,8,12,10,
3,7,8,0
```

INVADER

Once again it's time to defend the Earth as you shoot down a never ending stream of alien invaders. Allow nine to escape and the game ends. Press the space-bar to fire.

The memory locations in lines 160 are for the 48K Oric. If you have the 32K machine, these should be replaced with 30672 and 30703. For the 16K version use 14288 and 14319.

```
100 REM
110 REM  INVADER
120 REM
130 CLS
140 BS=9
150 PAPER 7:INK 2
155 REM  REDEFINE CHARACTERS
160 FOR J=47056 TO 47087
170 READ A:POKE J,A
180 NEXT J
190 DATA 0,30,63,45,63,12,18,33,0,0,12
,12,12,0,0,0
200 DATA 0,12,12,30,63,63,63,63,42,0,3
3,0,1,32,1,20
210 FOR J=1 TO 26
220 PRINT
230 NEXT J
240 PRINT CHR$(17)
250 B$="|":BX=19:BY=26
260 PLOT 0,26,4
272 PLOT BX,BY,124
280 PLOT 0,25,1
290 D$="SCORE: ":PLOT 6,0,D$

292 D$="BASES: "+STR$(BS):  PLOT 20,0,
D$
300 AY=INT(RND(1)*22)+2
310 AX=0:Z=0
320 PLOT 0,AY,2
```

```
324 PLOT AX,AY," "
330 AX=AX+1
332 IF AX=36 THEN BS=BS-1
334 IF BS=0 AND AX=36 THEN 600
336 IF AX=36 AND BS>0 THEN 290
340 PLOT AX,AY,"z"
350 K$=KEY$
360 IF K$=""OR Z=1 THEN 320
370 ZAP
372 Z=1
380 PLOT BX,BY-1,123
390 IF AX<>BX THEN PLOT BX,BY-1," ":GO
TO 320
400 PAPER 1
410 EXPLODE
420 PLOT AX,AY,125
430 SC=SC+10
440 D$=STR$(SC)
450 PLOT 13,0,D$
460 PAPER 7
470 PLOT AX,AY," "
480 PLOT BX,BY-1," "
490 GOTO 290
600 PRINT "YOUR SCORE = ";SC
610 PAPER 7:INK 0
620 PRINT CHR$(17)
630 END
```

MORSE CODE

This program produces random letters of morse code. An optional display of letters and a range of speeds is included.

At maximum speed it will generate approximately sixty characters per minute, which is adequate for the Radio Amateurs' morse test.

The duration of each letter can be adjusted by changing the value 8 in line 340.

```
100 REM
110 REM  MORSE CODE
120 REM
130 CLS
150 DIM M$(26)
160 REM  READ IN DATA
170 FOR J=1 TO 26
180 READ M$(J)
190 NEXT J
200 INPUT "SPEED (1 TO 500)";SP
210 PRINT:PRINT
220 IF SP>500 OR SP<1 THEN 200
230 INPUT "DISPLAY (Y/N)";Q$
240 PRINT
250 IF Q$="Y" THEN DI=1
270 REM  MAIN LOOP
280 R=INT(RND(1)*26)+1
290 IF DI=0 THEN 310
295 REM  DISPLAY LETTER
300 PRINT CHR$(R+64);"  ";
310 FOR J=1 TO LEN(M$(R))
320 PLAY 1,0,1,0
330 SOUND 1,100,9
340 WAIT 8*VAL(MID$(M$(R),J,1))
350 PLAY 0,0,0,0
360 WAIT 8
370 NEXT J
```

```
375 REM  SPEED DELAY
380 FOR D=1 TO 540-SP
390 NEXT D
400 GOTO 280
410 DATA 13,3111,3131,311,1,1131,331
420 DATA 1111,11,1333,313,1311,33
430 DATA 31,333,1331,3313,131,111,3
440 DATA 113,1113,133,3113,3133,3311
```

SIMON

This is your chance to see if your memory is better than your friend's.

Oric will print out a sequence of numbers for you to memorise. You will then be asked to type in your answer. If you are correct you move on to a longer sequence. However, if you are wrong, **Oric** will tell you the correct sequence – and won't you look silly!

```
100 REM
110 REM  SIMON
120 REM
130 CLS
140 CN=1
150 PRINT:PRINT
160 PRINT "YOUR SEQUENCE:":PRINT
170 R$=""
180 FOR I=1 TO CN
190 R=INT(RND(1)*9)
200 T$=STR$(R)
210 R$=R$+RIGHT$(T$,1)
220 NEXT I
230 PRINT R$
240 WAIT 200
250 CLS
260 PRINT:PRINT
270 INPUT "YOUR GUESS";G$
280 IF G$=R$ THEN PRINT "CORRECT!":CN=
CN+1:GOTO 350
300 PRINT "THE CORRECT SEQUENCE WAS:",
R$
350 INPUT "AGAIN";Q$
360 IF LEFT$(Q$,1)="N" THEN 380
370 GOTO 150
380 PRINT "YOU CORRECTLY REMEMBERED A
SEQUENCE OF";CN-1;"NUMBERS."
390 END
```

CHASE

This amusing game has you being chased around the screen by a strange looking monster to the accompaniment of some colourful sounds!

Guide yourself around the screen using the four arrow keys, but keep you fingers moving fast to survive.

The memory locations in line 150 are for the 48K Oric. If you have the 32K machine use 30672 and 30687. For the 16K use 14288 and 14303.

```
100 REM
110 REM  CHASE
120 REM
130 CLS
132 PRINT CHR$(17)
140 PAPER 7:INK 1
145 REM  DEFINE CHARACTERS
150 FOR J=47056 TO 47071
160 READ A
170 POKE J,A
180 NEXT J
190 DATA 28,20,28,30,41,20,18,33
200 DATA 30,43,63,55,63,33,63,0
205 REM  INITIAL VALUES
210 R=23:RO=23
220 C=16:CO=16
230 PLOT C,R,122
240 MR=2:MY=2
250 MC=16:MX=16
280 GOTO 740
290 REM  MOVE MAN
300 T$=KEY$
302 IF T$="" THEN 310
304 K$=T$
310 IF K$=CHR$(8) THEN C=C-1
320 IF K$=CHR$(9) THEN C=C+1
330 IF K$=CHR$(10) THEN R=R+1
340 IF K$=CHR$(11) THEN R=R-1
```

```
390 IF R=RO AND C=CO THEN 490
395 REM  EDGE OF SCREEN
400 IF R<1 OR R>26 OR C>35 OR C<1 THEN
 410 ELSE 440
410 R=RO:C=CO
420 GOTO 490
430 MUSIC 1,3,1,0
440 PLAY 1,0,1,300
450 PLOT CO,RO," "
460 PLOT C,R,122
470 RO=R
480 CO=C
490 RETURN
500 REM  MOVE MONSTER
510 IF MR<R THEN MR=MR+1
520 IF MR>R THEN MR=MR-1
530 IF MC<C THEN MC=MC+1
540 IF MC>C THEN MC=MC-1
580 MUSIC 1,2,5,0
590 PLAY 1,0,1,300
600 PLOT MX,MY," "
610 PLOT MC,MR,123
620 IF MR=R AND MC=C THEN 660
630 MY=MR
640 MX=MC
650 RETURN
655 REM  CAPTURED
660 FOR J=12 TO 1 STEP -1
670 MUSIC 1,3,J,0
680 PLAY 1,0,1,300
690 WAIT 10
692 PLAY 1,0,1,900
694 WAIT 30
700 NEXT J
702 MUSIC 1,3,1,0
704 PLAY 1,0,1,2000
706 WAIT 100
710 PLOT 12,12,"CAUGHT YOU!"
712 PRINT CHR$(17)
720 GOTO 780
```

```
730 REM  MAIN LOOP
740 GOSUB 300
750 GOSUB 510
760 GOSUB 300
770 GOTO 740
780 END
```

DIGITAL CLOCK

This program will display a working 24 hour digital clock in the centre of your television screen. As listed it keeps fairly good time but it can be tinkered with by changing the value of the wait statement in line 550 if desired. If the alarm function is not required, then enter any number greater than 23 for the hour when setting it.

Note that the z's in lines 310 to 350 and in line 512 are lower case. Press the CTRL and T keys together to change from capitals to lower case and vice-versa.

The memory locations in line 250 are for the 48K machine. If you have the 32K Oric use 30672 and 30679. For the 16K version use 14288 and 14295.

```
100 REM
110 REM  DIGITAL CLOCK
120 REM
130 CLS
140 PRINT
150 PRINT"PLEASE ENTER THE CORRECT TIM
E"
160 PRINT
170 INPUT "HOUR";HR
180 INPUT "MINUTES";MN
190 PRINT
200 PRINT "SET ALARM"
210 PRINT
220 INPUT "HOUR";AH
230 INPUT "MINUTE";AM
236 PRINT CHR$(17)
240 CLS
242 PAPER 7:INK 2
250 FOR J=47056 TO 47063
260 READ A
270 POKE J,A
280 NEXT J
290 DATA 63,63,63,63,63,63,63,63
300 REM  DRAW BOX
```

```
310 PLOT 14,12,"zzzzzzzzz"
320 PLOT 14,13,"z        z"
330 PLOT 14,14,"z        z"
340 PLOT 14,15,"z        z"
350 PLOT 14,16,"zzzzzzzzz"
356 GOSUB 630
360 REM  CALCULATE TIME
370 IF MN<=59 THEN 400
380 HR=HR+1
390 MN=0
392 GOSUB 630
400 IF HR<=23 THEN 420
410 HR=0
420 IF HR<10 AND MN<10 THEN 430 ELSE 4
50
430 T$="  "+STR$(HR)+":"+"0"+M$
440 GOTO 510
450 IF MN>9 THEN 480
460 T$=" "+STR$(HR)+":"+"0"+M$+" "
470 GOTO 510
480 IF HR>9 THEN 494
490 T$="  "+STR$(HR)+":"+M$
492 GOTO 510
494 T$=" "+STR$(HR)+":"+M$
500 REM  PRINT TIME
510 PLOT 14,14,T$
512 PLOT 14,14,"z"
520 REM  DELAY
530 FOR F=1 TO 60
540 PLOT 18,14," "
550 WAIT 97
560 PLOT 18,14,":"
580 IF HR<>AH OR MN<>AM THEN 610
590 MUSIC 1,3,3,0
600 PLAY 1,0,1,1000
610 NEXT F
620 MN=MN+1
622 GOSUB 630
624 GOTO 370
630 M$=STR$(MN)
```

```
640 M$=RIGHT$(M$,LEN(M$)-1)
650 RETURN
```

INKBLOTS

So you want to be a psychiatrist. Here is your chance to discover the dark corners of the mind.

This program will display a random pattern of inkblots similar to those used in the Rorschach inkblot test. Some strange and amusing shapes are often generated. Test yourself and your friends to see what **they** see in the shapes. You could get some very interesting results!

```
100 REM
110 REM  INKBLOTS
120 REM
130 HIRES
140 P=INT(RND(1)*6)+1
150 PAPER P
160 I=INT(RND(1)*6)+1
170 IF I=P THEN 160
180 INK I
190 A=INT(RND(1)*220)+3
200 B=INT(RND(1)*170)+3
210 C=C+1
220 IF C>7 THEN 300
230 R=1
240 CURSET A,B,3
250 CIRCLE R,1
260 R=R+1
270 IF A+R>235 OR B+R>193 THEN 190
280 IF A-R<9 OR B-R<9 THEN 190
290 GOTO 250
300 C=0
310 WAIT 1000
320 GOTO 130
```

WEATHER FORECAST

If you share the view that all weather forecasting is guess work and that the meteorological people make it up as they go along – why don't you do the same?

You can throw in hurricanes, tornadoes and floods if you wish, and you might even be right one day.

```
100 REM
110 REM  WEATHER FORECAST
120 REM
130 CLS
140 PRINT
150 REM  READ IN DATA
160 FOR J=1 TO 4
170 READ A$(J),B$(J),C$(J),D$(J),E$(J)

180 NEXT J
190 DEF FNR(X)=INT(RND(1)*X)+1
210 PRINT "GOOD EVENING.  HERE IS THE
WEATHER    FORECAST.  TONIGHT WILL BE
";
220 PRINT A$(FNR(4));"."
230 PRINT B$(FNR(4));
240 PRINT " OVERNIGHT WILL CLEAR, LEAV
ING TOMMOROW ";
250 PRINT A$(FNR(4));"."
252 PRINT
260 PRINT "THE WEST WILL BE ";
270 PRINT A$(FNR(4));
280 PRINT " WITH ";
290 PRINT C$(FNR(4));
300 PRINT " IN THE EAST."
302 PRINT
310 PRINT "FOR TRAVELLERS, MANCHESTER
AIRPORT IS ";
320 PRINT D$(FNR(4));
330 PRINT " AND ALL ROADS ARE ";
```

```
340 PRINT E$(FNA(4));"."
342 PRINT
350 PRINT "THE OUTLOOK IS ";
360 PRINT A$(FNA(4));"."
370 DATA FINE,SHOWERS,A DROUGHT,FOG BO
UND,SLIPPERY
380 DATA CLOUDY,SNOW,FALLING FROGS,FLO
ODED,ICY
390 DATA MILD,FROST,THUNDERSTORMS,COVE
RED BY FROGS,PASSABLE WITH CARE
400 DATA CANCELLED,A PLAGUE OF LOCUSTS
,BLIZZARDS,NO LONGER USED
410 DATA FULL OF POTHOLES
```

EGG TIMER

Here is a new version of breakfast television.

A working egg timer. You will be so fascinated with watching the sand run through your very own high technology egg timer that you will probably forget to take your egg off the boil. If you are the hard boiled type you should increase the value of the wait statement in line 425.

```
100 REM
110 REM  EGG TIMER
120 REM
130 HIRES
132 INK 3
135 REM  DRAW TIMER
140 CURSET 100,90,0
150 DRAW 40,0,1
160 DRAW 0,40,1
170 DRAW -15,0,1
180 DRAW -10,10,1
190 DRAW -15,0,1
200 DRAW 0,40,1
210 DRAW 40,0,1
220 DRAW 0,-40,1
230 DRAW -15,0,1
240 DRAW -10,-10,1
250 DRAW -15,0,1
260 DRAW 0,-40,1
265 REM  FILL TIMER
270 FOR J=100 TO 130
280 CURSET 100,J,1
290 DRAW 40,0,1
300 NEXT J
305 REM  FILL NECK
310 CURSET 116,131,1
320 DRAW 8,0,1
330 CURSET 118,132,1
340 DRAW 5,0,1
```

```
350 CURSET 119,133,1
360 DRAW 3,0,1
390 REM  MAIN LOOP
400 FOR J=100 TO 129
410 CURSET 101,J,0
420 DRAW 38,0,0
421 FOR K=1 TO 10
422 CURSET 120,140,1
423 CURSET 120,140,0
425 WAIT 54
426 CURSET 120,143,1
427 CURSET 120,143,0
428 NEXT K
430 CURSET 100,280-J,1
440 DRAW 40,0,1
450 NEXT J
460 PING
480 END
```

FIREWORK

Here is a crackerjack for your home entertainment.

You can save a fortune, keep warm inside and spare your pet's feelings by having your firework display indoors. Your high resolution screen will display a spluttering multi-coloured firework with accompanying sounds. You can even change the shape of the firework by playing around with lines 200 to 220.

If you want to change the sound effects try lines 270 and 272.

```
100 REM
110 REM  FIREWORK
120 REM
130 HIRES
135 REM  BLANK SCREEN
140 INK 0
145 REM  DRAW FIREWORK
150 CURSET 110,170,0
160 DRAW 0,-10,1
170 DRAW 8,0,1
180 DRAW 0,10,1
190 DRAW -8,0,1
198 FOR J=1 TO 50
200 R1=INT(RND(1)*160)+30
210 CURSET R1,20,0
220 DRAW -R1+114,140,1
230 NEXT J
235 REM  COLOURS
240 FOR J=1 TO 100
250 R2=INT(RND(1)*6)+1
260 INK R2
270 MUSIC 3,3,3,0
272 PLAY 0,1,2,1000
280 WAIT 15
290 NEXT J
295 REM  ERASE
300 FOR J=20 TO 159
```

```
310 CURSET 10,J,0
320 DRAW 200,0,0
330 NEXT J
334 IF N=1 THEN END
336 WAIT 500
338 REM  SURPRISE
340 CURSET 115,159,0
350 DRAW 0,-110,1
360 MUSIC 3,3,3,0
370 PLAY 0,1,1,1000
380 N=1
390 GOTO 300
```

MAZE CHASE

Maze Chase has you being chased around a maze by four evil looking blue ghosts intent on your downfall!

You can move around the maze by pressing keys O and P to move left and right and keys Q and A to move up and down. Points are gained just by staying alive.

You will need to keep your wits about you, as well as moving with lightning reactions, as this game needs some skill to play.

1500+ can be considered a respectable score.

The alarming rows of small z's in lines 690 to 930 form the structure of the maze, z being redefined in line 620 onwards as a solid block. The structure of the maze can thus be easily changed by typing in a different formation of z's.

The memory locations in line 620 are for the 48K **Oric** For the 32K machine replace them with 30672 and 30695. For the 16K version use 14288 and 14311.

```
100 REM
110 REM  MAZE CHASE
120 REM
130 GOTO 600
140 REM  MOVE
150 T$=KEY$
160 IF T$="" THEN 180
170 K$=T$
180 RN=RO
190 CN=CO
200 IF K$="O" THEN CN=CN-1
210 IF K$="P" THEN CN=CN+1
220 IF K$="A" THEN RN=RN+1
230 IF K$="Q" THEN RN=RN-1
240 IF CN>37 OR CN<1 OR RN>25 OR RN<1
THEN 350
250 X=SCRN(CN,RN)
```

```
260 IF X=122 THEN 350
270 IF X=124 THEN 1220
280 IF CN=CO AND RN=RO THEN 350
290 MUSIC 1,3,1,0
300 PLAY 1,0,1,500
310 PLOT CO,RO," "
320 PLOT CN,RN,123
330 RO=RN
340 CO=CN
350 RETURN
390 REM  MOVE GHOSTS
400 R=INT(RND(1)*4)+1
410 IF NS(R,1)=RN THEN 490
420 PLOT NS(R,2),NS(R,1),ST(R)
430 NS(R,1)=NS(R,1)+SGN(RN-NS(R,1))
440 STR(R)=SCRN(NS(R,2),NS(R,1))
450 IF STR(R)<>124 THEN 470
460 STR(R)=32
470 PLOT NS(R,2),NS(R,1),124
480 MUSIC 1,2,2,0:PLAY 1,0,1,500
490 IF NS(R,2)=CN THEN 560
500 PLOT NS(R,2),NS(R,1),STR(R)
510 NS(R,2)=NS(R,2)+SGN(CN-NS(R,2))
520 STR(R)=SCRN(NS(R,2),NS(R,1))
530 IF STR(R)<>124 THEN 550
540 STR(R)=32
550 PLOT NS(R,2),NS(R,1),124
560 IF RN=NS(R,1) AND CN=NS(R,2) THEN
1220
570 RETURN
590 REM  START
600 CLS
610 PAPER 7:INK 4
615 REM  DEFINE CHARACTERS
620 FOR J=47056 TO 47079
630 READ A
640 POKE J,A
650 NEXT J
660 DATA 63,63,63,63,63,63,63,63
670 DATA 28,20,28,9,62,40,20,20
680 DATA 12,30,45,63,51,30,30,63
```

```
685 REM  PRINT MAZE
690 PLOT 1,1,"zzzzzzzzzzzzzzzzzzzzzzz
zzzzzzzzzzzzz"
700 PLOT 1,2,"z
              z"
710 PLOT 1,3,"z zzzzzzzzz zzzzzzzzzz
zzz zzzzzzz z"
720 PLOT 1,4,"z
              z"
730 PLOT 1,5,"z zzzzzzzzz zzzzzzzzzz
zzz zzzzzzz z"
740 PLOT 1,6,"z
              z"

750 PLOT 1,7,"z zzzzzzzzz zzzzzzzzzz
zzz zzzzzzz z"
760 PLOT 1,8,"z
              z"
770 PLOT 1,9,"z zzzzzzzzz zzzzzzzzzz
zzz zzzzzzz z"
780 PLOT 1,10,"z
               z"
790 PLOT 1,11,"z zzzzzzzzz zzzzzzzzzz
zzzz zzzzzzz z"
800 PLOT 1,12,"z
               z"
810 PLOT 1,13,"z zzzzzzzzz zzzzzzzzzz
zzzz zzzzzzz z"
820 PLOT 1,14,"z
               z"
830 PLOT 1,15,"z zzzzzzzzz zzzzzzzzzz
zzzz zzzzzzz z"
840 PLOT 1,16,"z
               z"
850 PLOT 1,17,"z zzzzzzzzz zzzzzzzzzz
zzzz zzzzzzz z"
860 PLOT 1,18,"z
               z"
870 PLOT 1,19,"z zzzzzzzzz zzzzzzzzzz
zzzz zzzzzzz z"
880 PLOT 1,20,"z
               z"
```

```
890 PLOT 1,21,"z zzzzzzzzz zzzzzzzzz
zzzz zzzzzzzz z"
900 PLOT 1,22,"z
               z"
910 PLOT 1,23,"z zzzzzzzzz zzzzzzzzz
zzzz zzzzzzzz z"
920 PLOT 1,24,"z
               z"
930 PLOT 1,25,"zzzzzzzzzzzzzzzzzzzzz
zzzzzzzzzzzzzzz"
990 REM  INITIAL VALUES
1000 NS(1,1)=2:NS(1,2)=5
1010 NS(2,1)=2:NS(2,2)=30
1020 NS(3,1)=24:NS(3,2)=5
1030 NS(4,1)=24:NS(4,2)=30
1032 K$=CHR$(8)
1034 PRINT CHR$(17)
1040 PLOT 12,0,"SCORE = "
1050 FOR J=1 TO 4
1060 ST(J)=32
1070 PLOT NS(J,2),NS(J,1),124
1080 NEXT J
1082 FOR J=1 TO 25
1084 PRINT
1086 NEXT J
1090 RN=12:RO=12
1100 CN=18:CO=18
1110 PLOT CN,RN,123
1115 REM  MAIN LOOP
1120 GOSUB 400
1130 GOSUB 150
1140 GOSUB 400
1150 SC=SC+10
1160 S$=STR$(SC)
1170 S$=RIGHT$(S$,LEN(S$)-1)
1180 PLOT 20,0,S$
1190 GOTO 1120
1210 REM  CAUGHT
1220 FOR J=12 TO 1 STEP -1
1230 MUSIC 1,2,J,0
```

```
1240 PLAY 1,0,1,1000
1250 WAIT 20
1260 NEXT J
1270 INPUT "PLAY AGAIN";Q$
1280 IF LEFT$(Q$,1)="N" THEN 1300
1290 RUN
1300 PRINT CHR$(17)
1310 END
```

SNAP

Do you find that everytime you play Snap in your house there is an argument about who got their 'snap' down first?

Now you can let your **Oric** take the strain – as long as none of the players have hands like sledgehammers. A pair of symbols will appear on the screen. When they are the same, player one presses the Q key and player two presses the P key.

Oric will be the judge and award one point for the fastest response. The slower player of the two will lose a point. Cheating will also cost the culprit one point. The game ends when either player reaches 20 points.

```
100 REM
110 REM  SNAP
120 REM
130 CLS
140 PLOT 14,12,"S N A P !"
150 WAIT 500
160 CLS
165 PRINT CHR$(17)
170 PRINT:PRINT
175 REM  SELECT SYMBOLS
180 A=INT(RND(1)*10)+33
190 B=INT(RND(1)*10)+33
200 D$=CHR$(A)+"        "+CHR$(B)
202 MUSIC 1,2,9,4
204 PLAY 1,0,1,500
210 PLOT 14,12,D$
220 WAIT 20
222 PLAY 0,0,0,0
225 REM  LOOK AT KEYBOARD
230 K$=KEY$
240 IF K$<>"" THEN 260
250 IF A=B THEN 230
260 IF K$="Q" AND A=B THEN P1=P1+1
270 IF K$="P" AND A=B THEN P2=P2+1
```

```
280 IF K$="Q" AND A<>B THEN P1=P1-1
290 IF K$="P" AND A<>B THEN P2=P2-1
295 REM  PRINT SCORES
300 D$="PLAYER ONE: "+STR$(P1)+"   PLA
YER TWO: "+STR$(P2)
302 IF P1>19 OR P2>19 THEN 340
310 PLOT 2,2,D$
320 PLOT 1,2,2
330 GOTO 180
340 END
```

ENTRAPMENT

You will need to be very friendly with your partner to play this game. Each player has to control four keys to stay alive.

Both players control their own snake which winds it's way around the screen. If a snake hits any obstackle, including it's own body, or tries to leave the screen the player loses.

Player one uses keys E, S, D and X to move up, left, right and down.

Player two uses I, J, K and M as player one.

The memory locations in line 620 are for the 48K **Oric** the 32K model uses lines 30672 and 30687. The 16K model uses lines 14288 and 14303.

```
100 REM
110 REM  ENTRAPMENT
120 REM
130 GOTO 600
140 REM  PLAYER ONE
150 T$=KEY$
160 IF T$="" THEN 180
162 IF T$<>"E" AND T$<>"X" AND T$<>"S"
 AND T$<>"D" THEN 180
170 K$=T$
180 IF K$="E" THEN R1=R1-1
190 IF K$="X" THEN R1=R1+1
200 IF K$="S" THEN C1=C1-1
210 IF K$="D" THEN C1=C1+1
215 REM  EDGE OF SCREEN
220 IF R1>25 OR R1<1 OR C1>37 OR C1<1
 THEN 240
230 IF SCRN(C1,R1)=122 OR SCRN(C1,R1)=
123 THEN 240 ELSE 320
235 REM  CRASHED
240 FOR J=1 TO 20
250 MUSIC 1,3,6,0
260 PLAY 1,0,1,500
270 PLOT C1,R1," "
```

```
280 WAIT 10
290 PLOT C1,R1,122
300 NEXT J
310 RUN
320 PLOT C1,R1,122
350 RETURN
360 REM  PLAYER TWO
370 T$=KEY$
380 IF T$="" THEN 400
382 IF T$<>"I" AND T$<>"M" AND T$<>"J"
 AND T$<>"K" THEN 400
390 L$=T$
400 IF L$="I" THEN R2=R2-1
410 IF L$="M" THEN R2=R2+1
420 IF L$="J" THEN C2=C2-1
430 IF L$="K" THEN C2=C2+1
440 IF R2>25 OR R2<1 OR C2>37 OR C2<1
THEN 450
444 IF SCRN(C2,R2)=122 OR SCRN(C2,R2)=
123 THEN 450 ELSE 530
447 REM  CRASHED
450 FOR J=1 TO 20
460 MUSIC 1,4,6,0
470 PLAY 1,0,1,500
480 PLOT C2,R2," "
490 WAIT 10
500 PLOT C2,R2,123
510 NEXT J
520 RUN
530 PLOT C2,R2,123
540 RETURN
595 REM  START
600 CLS
601 REM  MOVE CURSOR
602 FOR M=1 TO 28
604 PRINT
606 NEXT M
610 PAPER 7:INK 4
615 REM  DEFINE CHARACTERS
620 FOR J=47056 TO 47071
630 READ A
```

```
640 POKE J,R
650 NEXT J
660 DATA 8,28,30,63,62,28,8,0
670 DATA 8,28,30,63,62,28,8,0
675 REM  INITIAL VALUES
680 R1=2
690 C1=4
700 K$="X"
710 R2=24
720 C2=37
730 L$="I"
735 REM  MAIN LOOP
740 GOSUB 370
750 GOSUB 150
760 GOTO 740
```

METEOR RUN

Caught in the middle of a meteor storm are a number of helpless aliens who need your help.

Using keys O and P to dodge the meteors which are hurtling towards your spaceship, you must pick up the aliens to gain extra points. The memory location in line 160 is for the 48K Oric. For the 32K version use 30672 and 30695. The 16K machine will use 14288 and 14311.

```
100 REM
110 REM  METEOR RUN
120 REM
130 CLS
135 REM  TURN OFF CURSOR
140 PRINT CHR$(17)
150 PLOT 12,12,"METEOR RUN"
152 WAIT 200
155 REM  DEFINE CHARACTERS
160 FOR J=47056 TO 47079
170 READ A:POKE J,A
180 NEXT J
190 DATA 12,30,62,63,63,31,30,12
200 DATA 12,45,45,63,45,45,12,30
210 DATA 0,30,63,45,63,12,18,33
215 REM  INITIAL VALUES
220 R=5
221 C=19
222 CLS
224 FOR M=1 TO 26
226 PRINT
228 NEXT M
230 PAPER 0:INK 7
260 RN=INT(RND(1)*38)+1
262 IF RND(1)>.9 THEN PLOT RN,26,"!":G
OTO 290
270 PLOT RN,26,"z"
280 SC=SC+5
282 IF SC/105=INT(SC/105) THEN PLOT C,
R," ":R=R+1
```

```
284 IF R>25 THEN R=25
290 PLOT C,R," "
300 PRINT
310 X=SCRN(C,R)
320 IF X<>124 THEN 360
325 REM  ALIEN
330 MUSIC 1,3,5,0
340 PLAY 1,0,1,200
350 SC=SC+50
352 GOTO 370
355 REM  COLLISION
360 IF SCRN(C,R)<>32 THEN 450
365 REM  LOOK AT KEYBOARD
370 K$=KEY$
380 IF K$="O" AND C>1 THEN C=C-1
390 IF K$="P" AND C<38 THEN C=C+1
400 PLOT C,R,"("
420 GOTO 260
440 REM  CRASHED
450 PAPER 1
460 EXPLODE
470 WAIT 100
472 PAPER 0
480 FOR M=12 TO 1 STEP -1
490 MUSIC 1,2,M,0
500 PLAY 1,0,5,100
502 WAIT 25
510 NEXT M
520 PLAY 0,0,0,0
530 PLOT 12,12,"G A M E    O V E R"
560 PRINT
570 IF SC>HS THEN HS=SC
580 PRINT "HIGH SCORE: ";HS
590 PRINT
600 PRINT "YOUR SCORE: ";SC
610 PRINT
620 INPUT "PLAY AGAIN";Q$
630 IF LEFT$(Q$,1)="N" THEN 680
640 SC=0
650 CLS
```

```
660 GOTO 220
680 PRINT CHR$(17)
690 PAPER 7:INK 0
700 END
```

BOMBER

You may have played a version of this game before in an arcade, but this time you only have **one** life!

As the pilot of a plane you are faced with a desperate fuel shortage. You are circling down towards a city which has been cleared for your crash. Now you have one last chance. Can you bomb away the buildings and clear a runway before it's too late? Release your bombs using the space bar. Be quick and good luck.

The memory locations in line 810 are for the 48K **Oric** If you have a 32K version use lines 30672 and 30703. The 16K model uses lines 14288 and 14319.

```
100 REM
110 REM  BOMBER
120 REM
150 GOTO 800
160 REM  MOVE PLANE
190 PLOT PX,PY,32
200 PX=PX+DR
205 REM  SCREEN EDGE?
210 IF PX>1 THEN 250
220 PC$=CHR$(122)
230 DR=1
240 PY=PY+1
245 REM  SCREEN EDGE?
250 IF PX<37 THEN290
260 PC$=CHR$(123)
270 DR=-1
280 PY=PY+1
290 IF SCRN(PX,PY)=124 THEN 1100
294 PLOT PX,PY,PC$
300 IF PX=36 AND PY=26 THEN 1260
350 RETURN
360 REM DROP BOMB
370 IF B=0 THEN 560
380 PLOT BX,BY," "
390 BX=BX+BD
```

```
400 BY=BY+2
405 REM  EDGE OF SCREEN
410 IF BY>26 OR BX<2 OR BX>36 THEN 510

420 IF SCRN(BX,BY)<>124 THEN 550
425 REM  BUILDING HIT
430 REPEAT
440 PAPER 1
450 EXPLODE:WAIT 5
460 PLOT BX,BY,125
470 BY=BY-1
480 PAPER 7
490 PLOT BX,BY+1," "
494 SC=SC+10
500 UNTIL SCRN(BX,BY)<>124
510 B=0
520 BX=0
530 BY=0
540 GOTO 560
550 PLOT BX,BY,":"
560 RETURN
600 REM  LOOK AT KEYBOARD
610 IF KEY$="" OR BL<1 OR B=1 THEN
 700
615 REM  DROP BOMB
620 B=1
630 BX=PX:BY=PY:BD=DR
640 BL=BL-1
680 MUSIC 1,3,5,0
690 PLAY 1,0,1,200
700 RETURN
710 REM PRINT SCORE ETC
716 S$=STR$(SC)
720 M$=RIGHT$(S$,LEN(S$)-1)
730 PLOT 12,0,M$
736 S$=STR$(BL)+" "
740 M$=RIGHT$(S$,LEN(S$)-1)
750 PLOT 27,0,M$
790 RETURN
800 CLS
802 FOR J=1 TO 27:PRINT:NEXT J
```

```
805 REM  DEFINE CHARACTERS
810 FOR J=47056 TO 47087
820 READ A:POKE J,A
830 NEXT J

840 DATA 0,0,40,62,63,8,0,0,0,0,5,31,6
3,4,0,0
850 DATA 63,45,63,45,63,45,63,63,42,21
,42,21,42,21,42,21
860 PRINT CHR$(17)
870 REM BUILD CITY
880 FOR J=2 TO 36
890 FOR K=INT(RND(1)*6)+20 TO 26
900 PLOT J,K,124
910 NEXT K
920 NEXT J
930 REM INITIAL POSITIONS
940 PX=34:PY=1:PC$=CHR$(123)
950 DR=-1:BL=75
952 REM  PRINT HEADINGS
956 S$=STR$(SC)
960 M$="SCORE:  "+RIGHT$(S$,LEN(S$)-1)
970 PLOT 5,0,M$
976 S$=STR$(BL)
980 M$="BOMBS:  "+RIGHT$(S$,LEN(S$)-1)
990 PLOT 20,0,M$
995 REM  MAIN LOOP
1000 GOSUB 190
1010 GOSUB 360
1020 GOSUB 600
1024 GOSUB 710
1026 WAIT 5
1030 GOTO 1000
1090 REM  CRASHED
1100 PAPER 1
1110 EXPLODE
1120 PAPER 7
1130 EXPLODE
1140 PLOT 12,12,"YOU CRASHED!"
1150 GOTO 1270
1190 REM LANDED
```

```
1260 PLOT 9,12,"A SUCCESSFUL LANDING!"

1270 PRINT CHR$(17)
1272 IF SC>HS THEN HS=SC
1273 PRINT
1274 PRINT "HIGH SCORE: ";HS
1276 PRINT
1280 PRINT:PRINT "YOUR SCORE: ";SC
1300 PRINT
1310 INPUT "PLAY AGAIN";Q$
1320 IF LEFT$(Q$,1)="N" THEN 1350
1330 SC=0
1332 RESTORE
1340 GOTO 800
1350 END
```

KALEIDOSCOPE

There are not many micro computers which can show off their colour capabilities in under 20 lines – but **Oric** can. Against a varying background Kaleidoscope will plot an ever changing series of colour blocks.

Although quite short this program will provide a fascinatingly hypnotic display of the colour range of the **Oric's** low resolution screen.

```
100 REM
105 REM  KALEIDOSCOPE
110 REM
120 CLS
125 PRINT CHR$(17)
130 R1=INT(RND(1)*38)
140 R2=INT(RND(1)*25)+1
150 PLOT R1,R2,254
160 R3=INT(RND(1)*36)
170 R4=INT(RND(1)*26)
180 R5=INT(RND(1)*7)+1
190 PLOT R3,R4,R5
195 IF RND(1)>.01 THEN 130
200 R6=INT(RND(1)*6)+1
210 INK R6
220 R7=INT(RND(1)*6)+1
230 PAPER R7
240 GOTO 130
```

ROAD RUNNER

The brakes on your car have failed and you are weaving down a rubble strewn road. Your objective is to stay on the road, avoid the rubble, and run up the highest mileage you can. We have given you a chance to notch up a few miles by letting you have three lives. You will need to be able to steer the car so we have also given you two keys which are defined at the beginning of the game.

```
100 REM
105 REM  ROAD RUNNER
110 REM
112 CLS
114 PRINT
116 INPUT "WHICH KEY FOR LEFT";L$
118 PRINT
120 INPUT "WHICH KEY FOR RIGHT";R$
122 CLS
130 GOTO 210
135 REM  PLOT ROAD
140 FOR J=1 TO R1
150 PLOT J,K,123
160 NEXT J
170 FOR J=R2 TO 35
180 PLOT J,K,123
190 NEXT J
192 R3=INT(RND(1)*5)+R1
194 PLOT R3,K,123
200 RETURN
205 REM  DEFINE CHARACTERS
210 FOR J=47056 TO 47071
220 READ A
230 POKE J,A
240 NEXT J
250 DATA 12,63,63,30,63,63,12,0
252 DATA 63,63,63,63,63,63,63,63
255 REM  INITIAL VALUES
260 PAPER 7:INK 0
```

```
270 R1=16
272 R2=21
280 C=18
282 R=26
284 PRINT CHR$(17)
290 FOR K=1 TO 26
300 GOSUB 140
310 NEXT K
320 K=1
330 LV=3
335 REM SCROLL
340 PRINT CHR$(11);
345 REM  LOOK AT KEYBOARD
350 K$=KEY$
360 IF K$=L$ THEN C=C-1
370 IF K$=R$ THEN C=C+1
380 IF SCRN(C,R)=32 THEN 430
385 REM  COLLISION
390 EXPLODE
400 LV=LV-1
410 IF LV=0 THEN 500
415 REM  SAFE POSITION
420 IF K$=L$ THEN C=C+1
422 IF K$=R$ THEN C=C-1
430 PLOT C,R,122
440 GOSUB 140
450 R1=R1+INT(RND(1)*3)-1
460 R2=R1+5
470 IF R1<2 OR R2>35 THEN 450
480 SC=SC+1
490 GOTO 340
495 REM  END ROUTINE
500 FOR M=1 TO 28
510 PRINT
520 NEXT M
530 PRINT "TOTAL MILEAGE = ";SC
540 PRINT CHR$(17)
550 END
```

TOWERS OF HANOI

Here is an opportunity to prove you know your numbers. Towers of Hanoi is a simulation of the well known puzzle where you have to move a column of numbers from the left hand tower (No. 1) to the right hand tower (No. 3), in as few moves as possible.

Seems simple, but remember that you can't place a larger number on top of a smaller number. **Oric** will ask you which number you wish to move and to where. Get it right and congratulations are in order – get it wrong and you will get that awful message "you can't do that".

```
100 REM
102 REM  TOWERS OF HANOI
104 REM
106 CLS:PRINT
108 CN=1
109 DIM A(3,11)
110 INPUT "NUMBER OF LEVELS (1 TO 10)"
;ND
112 IF ND<1 OR ND>10 THEN 110
114 FOR I=11-ND TO 10
116 A(1,I)=I
118 NEXT I
119 FOR D=1 TO 20:PRINT:NEXT D
120 REM  PRINT TOWERS
122 FOR J=1 TO 10
124 FOR M=1 TO 3
126 IF A(M,J)=0 THEN 130
128 PLOT M*5,J+15,STR$(A(M,J))
130 NEXT M
132 PRINT
134 NEXT J
136 FOR I=11-ND TO 10
138 IF A(3,I)<>I THEN 146
140 NEXT I
142 PRINT
144 PRINT "CONGRATULATIONS!  COMPLETED
 IN ONLY ";NM;" MOVES":END
```

```
146 PRINT
148 NM=NM+1
150 INPUT "MOVE FROM WHICH TOWER";K
152 PRINT
154 REM  VALIDATE INPUT
156 IF K<1 OR K>3 THEN 150
158 FOR I=1 TO 10
160 IF A(K,I)=0 THEN 168
162 TH=A(K,I)
164 A(K,I)=0
166 GOTO 172
168 NEXT I
170 GOTO 150
172 INPUT "TO WHERE";K
174 PRINT
176 REM VALIDATE INPUT
178 IF K<1 OR K>3 THEN 172
180 FOR I=1 TO 10
182 IF A(K,I)=0 THEN 190
184 IF A(K,I)<TH THEN PRINT "YOU CAN'T
 DO THAT!":GOTO 172
186 A(K,I-1)=TH
188 GOTO 122
190 NEXT I
192 A(K,10)=TH
194 GOTO 122
```

PROGRAMS TO MAKE YOU THINK

The following selection of programs is designed to show off your talents.

GUESS THE NUMBER

Here is a program to give the younger members of the family a chance to shine.

Oric will choose a number between one and a hundred. The player will then try and guess the number chosen. If, and it's fairly likely, the first guess is wrong **Oric** will say "too high" or "too low".

Once the player has got the high and low range they ought to be able to 'squeeze' both ends until they reach the number. **Oric** will then tell them how many guesses they needed. Kind parents can, of course, begin the program by setting the upper number at twenty.

```
100 REM
110 REM  GUESS THE NUMBER
120 REM
130 CLS
140 PRINT
150 PRINT "HELLO!":PRINT
160 PRINT "THIS IS A NUMBER GUESSING G
AME. I'M"
170 PRINT "GOING TO THINK OF A NUMBER
BETWEEN"
180 PRINT "1 AND 100.":PRINT
190 PRINT "YOU TRY AND GUESS WHAT IT I
S."
194 WAIT 500
200 PRINT
210 PRINT "THINKING....."
220 PRINT
230 WAIT 500
240 N=INT(RND(1)*100)+1
246 PRINT
250 INPUT "WHAT'S YOUR GUESS";G
252 TR=TR+1
260 IF G>N THEN PRINT "TOO HIGH! TRY A
 LOWER NUMBER.":GOTO 246
```

```
270 IF G<N THEN PRINT "TOO LOW! TRY A
HIGHER NUMBER.":GOTO 246
280 PRINT "CORRECT!":PRINT
290 PRINT "YOU GUESSED IN ";TR;"TRIES.
"
292 PRINT
300 TR=0
310 GOTO 210
```

LETTER

Alright, you may know how many letters there are in the alphabet (26 actually) but could you tell the difference, in numbers, between D and R?

Oric will print two letters on the screen. Press the key corresponding to the difference between the letters and see what happens. The loop between 200 and 250 gives a time limit with accompanying sound effects. The time limit can be changed via the wait statement in line 232.

```
100 REM
110 REM  LETTER
120 REM
130 CLS
140 PRINT:PRINT
145 REM  SELECT LETTERS
150 L=INT(RND(1)*26)+65
160 M=INT(RND(1)*26)+65
170 D=ABS(M-L)
180 IF D>9 THEN 150
190 PRINT CHR$(L);"   ";CHR$(M)
200 FOR J=12 TO 1 STEP -1
210 MUSIC 1,3,J,0
220 PLAY 1,0,1,500
225 REM  LOOK AT KEYBOARD
230 K$=KEY$
232 WAIT 50
240 IF VAL(K$)=D THEN 300
250 NEXT J
260 PRINT
262 MUSIC 1,1,1,0
264 PLAY 1,0,2,1000
270 PRINT "DIFFERENCE = ";D
280 WAIT 100
290 GOTO 140
300 PRINT "CORRECT!"
310 GOTO 280
```

CURRENCY CONVERTER

If you are planning a holiday trip or, more importantly, a business visit and wonder how far your money is going to go – here is the answer.

You type in the present rate from your newspaper and you can also predict a change in the rate and see what will happen to your hard earned pennies.

This example shows dollars and pounds but, of course, any currency can be used.

```
100 REM
110 REM  CURRENCY CONVERTER
120 REM
130 CLS
140 PRINT:PRINT
150 INPUT "CURRENT RATE $/POUND";R
160 PRINT
170 PRINT "1.  $ TO POUNDS"
180 PRINT "2.  POUNDS TO $"
190 PRINT
200 INPUT "1 OR 2";C
210 IF C<>1 AND C<>2 THEN 200
220 PRINT
230 ON C GOTO 240,300
240 INPUT D
250 PRINT INT((D/R)*100)/100
260 PRINT
270 GOTO 240
300 INPUT P
310 PRINT INT((P*R)*100)/100
320 PRINT
330 GOTO 300
```

ARITHMETIC

An arithmetic test for all ages.

This program will generate an endless sequence of addition and subtraction sums, testing for the correct answer in each case.

To stop the program (all good things must come to an end!) press the CTRL and C keys simultaneously.

```
100 REM
110 REM  ARITHMETIC
120 REM
130 CLS
140 FOR J=1 TO 20
150 PRINT
160 NEXT J
170 PLOT 12,8,"ARITHMETIC"
180 INPUT "MAXIMUM NUMBER";MN
190 CLS
200 PRINT:PRINT
204 CN=0
207 REM  PICK NUMBERS
210 R1=INT(RND(1)*MN)+1
220 R2=INT(RND(1)*MN)+1
225 REM  NO NEGATIVE NUMBERS
230 IF R2>R1 THEN 220
240 R3=INT(RND(1)*2)+1
245 REM  PRINT SUM
250 PRINT "WHAT IS ";R1;
260 IF R3=1 THEN PRINT "MINUS ";
264 IF R3<>1 THEN PRINT "PLUS ";
270 PRINT R2
275 REM  INPUT ANSWER
280 INPUT AN
282 PRINT
290 IF R3=1 THEN AS=R1-R2
294 IF R3<>1 THEN AS=R1+R2
300 IF AN<>AS THEN 310
302 MUSIC 1,4,7,0
```

```
304 PLAY 1,0,1,2000
306 PRINT "CORRECT!"
308 GOTO 200
310 CN=CN+1
312 MUSIC 1,2,3,0
314 PLAY 1,0,1,2000
315 REM  TOO MANY TRIES
320 IF CN=4 THEN PRINT "THE CORRECT AN
SWER IS ";A$:GOTO 200
330 PRINT "TRY AGAIN!"
340 GOTO 280
```

MULTIPLICATION TABLES

Here's a simple program to test your knowledge of multiplication. The range of numbers generated is controlled by an input statement at the beginning, which makes this program suitable for a range of ages and abilities. (Try calculating 397 x 422 in your head!)

```
100 REM
110 REM  MULTIPLICATION TABLES
120 REM
130 CLS
140 PRINT
150 INPUT "MAXIMUM NUMBER";MX
160 N1=INT(RND(1)*MX)+1
170 N2=INT(RND(1)*MX)+1
175 REM  ANSWER COUNT = 0
180 CT=0
190 PRINT
192 PRINT "WHAT IS ";N1;"TIMES ";N2;
200 INPUT AN
210 IF AN=N1*N2 THEN 260
215 REM  WRONG ANSWER
220 CT=CT+1
230 IF CT<3 THEN 240
232 PRINT
234 PRINT N1;"X ";N2;"= ";N1*N2
236 GOTO 160
240 PRINT
242 PRINT "TRY AGAIN!"
244 PRINT
250 GOTO 200
255 REM  RIGHT ANSWER
260 PRINT
262 PRINT "CORRECT!"
264 PRINT
270 INPUT "AGAIN";Q$
280 IF LEFT$(Q$,1)="N" THEN 300
290 GOTO 160
300 END
```

SPELLING TEST

Spelling Test will print out a list of three different spellings of a word, only one of which is correct. You will then be asked to enter the correct spelling. If your answer is correct a new word is given.

The list of words in the data statements can be varied to suit all ages. A modification would be to make none of the spellings correct!

```
100 REM
110 REM  SPELLING TEST
120 REM
130 CLS
140 PAPER 7:INK 0
150 DIM WD$(20,3)
160 PRINT CHR$(17)
170 PLOT 12,12,"SPELLING TEST"
180 WAIT 200
185 REM  READ IN DATA
190 FOR J=1 TO 20
200 FOR K=1 TO 3
210 READ WD$(J,K)
220 NEXT K,J
230 PRINT CHR$(17)
235 REM  SELECT WORD
240 R1=INT(RND(1)*3)+1
250 R2=INT(RND(1)*3)+1
260 IF R1=R2 THEN 250
270 R3=INT(RND(1)*3)+1
280 IF R3=R1 OR R3=R2 THEN 270
290 RD=INT(RND(1)*20)+1
300 IF RD=RWD THEN 290
310 RWD=RD
320 CLS:PRINT:PRINT
325 REM  PRINT CHOICE
330 PRINT "1. ";WD$(RWD,R1)
340 PRINT "2. ";WD$(RWD,R2)
350 PRINT "3. ";WD$(RWD,R3)
```

```
360 PRINT:PRINT
370 PRINT"WHAT IS THE CORRECT SPELLING
"
374 PRINT
380 INPUT AN$
390 PRINT
400 IF AN$=WD$(RWD,1) THEN 430
410 PRINT "TRY AGAIN!" :PRINT
420 GOTO 380
430 PRINT "CORRECT!"
440 WAIT 100:GOTO 240
450 DATA CARROT,CAROT,CAROTT
460 DATA TELEVISION,TELIVISION,TELEVIS
ON
470 DATA COMPUTER,COMPUTOR,COMPUTAR
480 DATA VEHICLE,VEICLE,VEERCLE
490 DATA RANDOM,RANDEM,RANDAM
500 DATA FEBRUARY,FEBUARY,FEBURARY
510 DATA YELLOW,YELOW,YELLEW
520 DATA SENTENCE,SENTANCE,SENTENSE
530 DATA SUCCESS,SUCESS,SUCCES
540 DATA TOMORROW,TOMMOROW,TOMOROW
550 DATA ENVIRONMENT,ENVIROMENT,ENVIRA
MENT
560 DATA BEAUTIFUL,BUEATIFUL,BEAUTIFUL
L
570 DATA IMPOSSIBLE,IMPOSIBLE,INPOSSIB
LE
580 DATA ILLEGIBLE,ILEGIBLE,ILLEGABLE
590 DATA AMEND,AMMEND,EMEND
600 DATA ORCHESTRA,ORKESTRA,ORCESTRA
610 DATA INTRIGUE,INTRIGE,INTREAGUE
620 DATA PNEUMATIC,NEWMATIC,PNUMATIC
630 DATA WEDNESDAY,WENSDAY,WENDSDAY
640 DATA APPLICATION,APLICATION,APPLIC
CATION
```

METRIC CONVERTER

Are you still having arguments in your house about the difference between chains, poles and metres and hectares? Here at last is a simple to use conversion program so that everyone in the family can be right.

Further routines can be added to cover different conversions for volumes etc.

```
100 REM
110 REM  METRIC CONVERTER
120 REM
130 CLS
140 PRINT:PRINT
150 PRINT "1. KILOGRAMS TO POUNDS"
160 PRINT "2. METRES TO FEET AND INCHE
S"
170 PRINT
180 INPUT"INPUT 1 OR 2 TO SELECT";KY
190 IF KY<>1 AND KY<>2 THEN 180
200 ON KY GOTO 210,320
210 PRINT:PRINT"INPUT METRIC":PRINT
220 INPUT KG
230 IF KG=0 THEN 430
240 OZ=KG*2.2046*16
250 IF INT(OZ)<16 THEN 280
260 PD=INT(OZ/16)
270 OZ=OZ-PD*16
280 PRINT PD;"LBS ";OZ;"OZS" :PRINT
290 PD=0
300 OZ=0
310 GOTO 220
320 PRINT:PRINT"INPUT METRIC":PRINT
330 INPUT MT
340 IF MT=0 THEN 430
350 IN=MT*39.37
360 IF INT(IN)<12 THEN 390
370 FT=INT(IN/12)
380 IN=IN-FT*12
```

```
390 PRINT FT;"FT ";IN;"IN":PRINT
400 FT=0
410 IN=0
420 GOTO 330
430 END
```

CAPITALS

This program will ask you for the capital of a country. Enter the right answer, spelt correctly, and a further question will be asked. If an incorrect answer is given, you will have three tries to get it right, after which the answer will be printed.

The list of countries and capitals can be extended by adding extra data statements after line 540 and increasing the numbers in lines 140, 150 and 190 to match the total number of countries.

This program can easily be altered to ask for information on counties or American state capitals etc. It can also be adapted to place people at places or dates in time, or to show the inventors or painters of a period or place.

```
100 REM
110 REM  CAPITALS
120 REM
130 CLS
140 DIM C$(20),P$(20)
150 FOR J=1 TO 20
160 READ C$(J),P$(J)
170 NEXT J
180 CN=0
190 RN=INT(RND(1)*20)+1
200 IF RN=R THEN 190
210 R=RN
220 PRINT
230 PRINT "WHAT IS THE CAPITAL OF ";C$
(R):PRINT
240 INPUT AN$:PRINT
250 IF AN$=P$(R) THEN PRINT "CORRECT!"
:GOTO 180
260 CN=CN+1
270 IF CN=4 THEN 310
280 PRINT "TRY AGAIN!"
290 PRINT
300 GOTO 240
310 PRINT
```

```
320 PRINT "THE CAPITAL OF ";C$(R);" IS
 ";P$(R);"."
330 PRINT
340 GOTO 180
350 DATA AFGHANISTAN,KABUL
360 DATA ARGENTINA,BUENOS AIRES
370 DATA BELGIUM,BRUSSELS
380 DATA BRAZIL,RIO DE JANEIRO
390 DATA CANADA,OTTOWA
400 DATA CUBA,HAVANA
410 DATA DENMARK,COPENHAGEN
420 DATA EGYPT,CAIRO
430 DATA ENGLAND,LONDON
440 DATA FRANCE,PARIS
450 DATA GREECE,ATHENS
460 DATA ITALY,ROME
470 DATA JAMAICA,KINGSTON
480 DATA NIGERIA,LAGOS
490 DATA NORWAY,OSLO
500 DATA PORTUGAL,LISBON
510 DATA RUSSIA,MOSCOW
520 DATA SCOTLAND,EDINBURGH
530 DATA SPAIN,MADRID
540 DATA U.S.A.,WASHINGTON
```

GRAPH IT

One of the horrors in business and education is being asked to plot a graph.

At this time it is unlikely that you can smuggle your **Oric** into a classroom, or some boardrooms, but you can practise plotting graphs with this easy to use program.

You, as user, can insert the function in line 130 of the program. The *40 in this example is a scaling factor which can be found either by calculating a few likely values, or by experiment. Try substituting the following in line 130

DEF FNA(X)=COS(X)*40

```
100 REM
110 REM  GRAPH IT
120 REM
125 REM  INSERT FUNCTION BELOW
130 DEF FNA(X)=SIN(X)*40
140 HIRES
145 REM  DRAW AXES
150 DRAW 0,199,1
160 CURSET 0,100,0
170 DRAW 238,0,1
175 REM  PLOT FUNCTION
180 FOR X=0 TO 11 STEP 0.01
190 A=-FNA(X)+100
200 IF A>199 OR A<0 THEN 230
220 CURSET X*20,A,1
230 NEXT X
```

REPORTS

Ever been stuck for an impressive sounding phrase for incorporation into reports or speeches? This program is the answer! Readers might like to insert their own words into the data statements in lines 270 – 330.

A similar structure to that of this program is used in many of the "poetry" generating programs around. Just insert your own short phrases instead of the single words.

```
100 REM
110 REM REPORTS
120 REM
130 CLS
140 REM  READ IN DATA
150 FOR J=1 TO 7
160 READ A$(J),B$(J),C$(J)
170 NEXT J
180 PRINT:PRINT
190 INPUT "HOW MANY PHRASES";PH
200 PRINT:PRINT
210 REM  PRINT PH PHRASES
220 FOR J=1 TO PH
230 PRINT A$(INT(RND(1)*7+1));" ";
240 PRINT B$(INT(RND(1)*7+1));" ";
250 PRINT C$(INT(RND(1)*7+1))
254 PRINT:PRINT
260 NEXT J
270 DATA ON-GOING,HUMANISTIC,ENVIRONME
NT
280 DATA MODULAR,INTEGRATED,SITUATION
290 DATA DIFFERENTIATED,MOTIVATIONAL,P
ROCESS
300 DATA MINORITY,CREATIVE,EXPERIMENT
310 DATA OPEN-ENDED,VERTICAL,TECHNIQUE

320 DATA INDIVIDUAL,EVALUATIVE,RESOURC
E
330 DATA HOMOGENEOUS,OBJECTIVE,FACILIT
Y
```

CHEQUE BOOK

This is a cheque book balancer to help you to keep track of your account. To use, enter your last known balance, followed by a list of cheques written out and payments in. Each list should be ended with a zero. The computer will prompt you for each list.

Line 270 is of interest – it uses the escape character CHR$ (27) followed by a control character "L" in front of the print string to flash the word "overdrawn" on and off.

Although presented for a simple cheque account, the program could easily be modified to cover many household accounting requirements.

```
100 REM
110 REM  CHEQUE BOOK
120 REM
130 CLS
132 PRINT
134 INPUT "HOW MANY ACCOUNTS";AC
136 PRINT
138 FOR L=1 TO AC
140 PRINT
144 PRINT "ACCOUNT NO. ";L
146 PRINT
150 INPUT "YOUR LAST BALANCE";BL
160 PRINT
162 PRINT "CHEQUES SINCE"
164 PRINT
170 INPUT CH
180 IF CH=0 THEN 210
190 BL=BL-CH
200 GOTO 170
210 PRINT
212 PRINT "PAYMENTS IN"
214 PRINT
220 INPUT PY
230 IF PY=0 THEN 260
240 BL=BL+PY
```

```
250 GOTO 220
260 PRINT
262 PRINT "YOUR CURRENT BALANCE IS ";B
L;
270 IF BL<0 THEN PRINT CHR$(27);"LOVER
DRAWN"
280 PRINT:PRINT
282 TB=TB+BL
284 BL=0
286 NEXT L
288 PRINT "TOTAL BALANCES = ";TB
290 PRINT
300 END
```

PROGRAMS FOR BUDDING PROGRAMMERS

This last selection of programs is designed to help those readers who want to stretch their Oric – and themselves.

MACHINE CODE MONITOR

This program is for readers who are ready for advanced programming, and who want to increase the speed of execution on their **Oric**. Using this program you can input machine code directly into **Oric's** memory, in either decimal or hexadecimal. You will find in this book a program to help you with binary/decimal conversion.

Even if you are experienced it is always worthwhile remembering to make sure your starting point precedes a vacant part of your machine's memory – and is available on your model. After all, there is no point in trying to put several K of changes into your machine if it's too small. To end your input type "S".

```
100 REM
110 REM  MACHINE CODE MONITOR
120 REM
130 INPUT "HEX OR DECIMAL INPUT";Q$
140 IF LEFT$(Q$,1)="D" THEN 260
150 INPUT "START LOCATION";ST
160 GOTO 180
170 ST=ST+1
180 PRINT ST;
190 INPUT CD$
200 IF CD$="S" THEN 350
210 CD$=RIGHT$(CD$,2)
220 CD$="#"+CD$
230 C=VAL(CD$)
240 POKE ST,C
250 GOTO 170
260 INPUT "START LOCATION";ST
270 GOTO 290
280 ST=ST+1
290 PRINT ST;
300 INPUT CD$
310 IF CD$="S" THEN 350
```

```
320 C=VAL(CD$)
330 POKE ST,C
340 GOTO 280
350 END
```

MEMORY PEEK

This, for the budding programmers amongst you, is a shortcut to understanding the operation of the ROM part of your machine's memory. The display of the memory content is in both decimal and hexadecimal.

Along with the preceding machine code monitor program this 'peek' will help you form a powerful machine code aid.

```
100 REM
110 REM  MEMORY PEEK
120 REM
130 CLS
140 PRINT:PRINT
150 INPUT "START ADDRESS";ST
160 PRINT
170 INPUT "END ADDRESS";EN
180 CLS
190 FOR J=ST TO EN STEP 20
200 PRINT
210 PRINT "ADDRESS       HEX       DECIM
AL"
220 PRINT
230 FOR K=J TO J+20
240 PRINT K;"        ";HEX$(PEEK(K));"
      ";PEEK(K)
250 NEXT K
260 GET A$
270 CLS
280 NEXT J
```

BINARY TO DECIMAL CONVERTER

This program will convert a sequence of binary digits (0's and 1's) into a decimal number. The length of the binary number is limited only by the maximum length of the input statement on your **Oric**!

```
100 REM
110 REM  BINARY TO DECIMAL CONVERSION
120 REM
130 CLS
140 PRINT
150 PRINT "INPUT YOUR BINARY NUMBERS A
S A STRING OF 1'S AND 0'S"
170 PRINT:PRINT "EG 100110"
180 PRINT:PRINT "INPUT S TO STOP"
185 PRINT
190 REM MAIN LOOP
200 INPUT "BINARY ";B$
220 IF B$="S" THEN 340
230 N=1
240 DC=0
250 FOR J=1 TO LEN(B$)
260 A$=MID$(B$,LEN(B$)-J+1,1)
270 IF A$="0" THEN 290
280 DC=DC+N
290 N=N*2
300 NEXT J
310 REM  PRINT DECIMAL
320 PRINT "DECIMAL =";DC
325 PRINT
330 GOTO 200
340 END
```

CHARACTER DEFINITION

This program will allow you to redefine any of the character set of your **Oric**. Input the character required followed by the new definition as eight rows of six binary digits. If you have a number of characters to do, it's a good idea to redefine zero as a blank space (enter 8 rows of 6 zeros) and one as a solid block (enter 8 rows of 6 ones). This will enable you to design each character in magnified form as you enter it.

The memory location in line 190 refers to the 48K Oric, for the 32K version, use 29696. For the 16K machine use 13312.

```
100 REM
110 REM  CHARACTER DEFINITION
120 REM
130 CLS:PRINT
140 INPUT "WHICH CHARACTER";C$
150 T=ASC(C$)
160 PRINT
170 PRINT "INPUT 8 ROWS AS A 6 DIGIT B
INARY       NUMBER"
180 PRINT
190 FOR K=46080+T*8 TO 46080+T*8+7
200 INPUT B$
202 IF LEN(B$)<>6 THEN PRINT "WRONG LE
NGTH!":GOTO 200
210 N=1
220 DC=0
230 FOR J=1 TO LEN(B$)
240 A$=MID$(B$,LEN(B$)-J+1,1)
250 IF A$="0" THEN 270
260 DC=DC+N
270 N=N*2
280 NEXT J
290 POKE K,DC
300 NEXT K
```

```
304 PRINT CHR$(T)
306 PRINT
310 FOR L=1 TO 20
320 PRINT CHR$(T);
330 NEXT L
340 PRINT
350 END
```

SCREEN PRINT

Screen Print will give a copy of the television display onto a line printer connected to the Centronics output port. Redefined characters will normally be printed in their standard undefined form. Different parts of the screen can be copied by changing the loop values in line 130 (number of lines) and line 140 (number of columns).

On some printers the listing will be double spaced. Single spacing can usually be achieved by replacing line 180 with LPRINT CHR$ (27)

```
100 REM
110 REM  SCREEN PRINT
120 REM
130 FOR J=0 TO 26
140 FOR K=0 TO 38
150 X=SCRN(K,J)
160 LPRINT CHR$(X);
170 NEXT K
180 LPRINT
190 NEXT J
200 END
```

50 programs for your ORIC–1

Here is a mammoth collection of high quality programs for all current versions of the ORIC–1

You can
run an obstacle course
protect six cities from missile attack
save a maiden from the dark knight
Using High Resolution Graphics
You can
draw graphs
plot circles
weave webs
even have your own firework display
Budding Programmers can
peek into the ORIC memory
redefine characters
input machine codes

This is the ORIC–1 program book
with something for everyone!!

ISBN 0 9465 7601 7
Phoenix Publishing Associates

£5.95

THE ORIC-1 PROGRAM BOOK V. APPS

PHOENIX
PUBLISHING ASSOCIATES